
Abstract
The XMod package provides functions for computation with

• finite crossed modules of groups and cat1-groups, and morphisms of these structures;

• finite pre-crossed modules, pre-cat1-groups, and their Peiffer quotients;

• isoclinism classes of groups and crossed modules;

• derivations of crossed modules and sections of cat1-groups;

• crossed squares and their morphisms, including the actor crossed square of a crossed module;

• crossed modules of finite groupoids (experimental version).

XMod was originally implemented in 1996 using the GAP3 language, when the second author was studying
for a Ph.D. [Alp97] in Bangor.

In April 2002 the first and third parts were converted to GAP4, the pre-structures were added, and version
2.001 was released. The final two parts, covering derivations, sections and actors, were included in the January
2004 release 2.002 for GAP 4.4.

In October 2015 functions for computing isoclinism classes of crossed modules, written by Alper Odabaş
and Enver Uslu, were added. These are contained in Chapter 4, and are described in detail in the paper [IOU16].

Bug reports, suggestions and comments are, of course, welcome. Please submit an is-
sue at http://github.com/gap-packages/xmod/issues/ or send an email to the first author at
c.d.wensley@bangor.ac.uk.

Copyright
© 1996-2018, Chris Wensley et al.

The XMod package is free software; you can redistribute it and/or modify it under the terms of the GNU
General Public License as published by the Free Software Foundation; either version 2 of the License, or (at
your option) any later version.

Chapter 1

Introduction

The XMod package provides functions for computation with

• finite crossed modules of groups and cat1-groups, and morphisms of these structures;

• finite pre-crossed modules, pre-cat1-groups, and their Peiffer quotients;

• derivations of crossed modules and sections of cat1-groups;

• isoclinism of groups and crossed modules;

• the actor crossed square of a crossed module;

• crossed squares, cat2-groups, and their morphisms (experimental version);

• crossed modules of groupoids (experimental version).

It is loaded with the command
Example

gap> LoadPackage( "xmod" );

The term crossed module was introduced by J. H. C. Whitehead in [Whi48], [Whi49]. Loday,
in [Lod82], reformulated the notion of a crossed module as a cat1-group. Norrie [Nor90], [Nor87]
and Gilbert [Gil90] have studied derivations, automorphisms of crossed modules and the actor of
a crossed module, while Ellis [Ell84] has investigated higher dimensional analogues. Properties of
induced crossed modules have been determined by Brown, Higgins and Wensley in [BH78], [BW95]
and [BW96]. For further references see [AW00], where we discuss some of the data structures and
algorithms used in this package, and also tabulate isomorphism classes of cat1-groups up to size 30.

XMod was originally implemented in 1997 using the GAP 3 language. In April 2002 the first and
third parts were converted to GAP 4, the pre-structures were added, and version 2.001 was released.
The final two parts, covering derivations, sections and actors, were included in the January 2004
release 2.002 for GAP 4.4. Many of the function names have been changed during the conversion, for
example ConjugationXMod has become XModByNormalSubgroup (2.1.1). For a list of name changes
see the file names.pdf in the doc directory.

In October 2015 Alper Odabaş and Enver Uslu were added to the list of package authors. Their
functions for computing isoclinism classes of groups and crossed modules are contained in Chapter 4,
and are described in detail in their paper [IOU16].

5

The package may be obtained as a compressed tar file XMod-version.number.tar.gz by ftp
from one of the following sites:

• the XMod GitHub release site: https://github.com/gap-packages.github.io/xmod/.

• any GAP archive, e.g. https://www.gap-system.org/Packages/packages.html;

The package also has a GitHub repository at: https://github.com/gap-packages/xmod/.
Crossed modules and cat1-groups are special types of 2-dimensional groups [Bro82], [BHS11],

and are implemented as 2DimensionalDomains and 2DimensionalGroups having a Source and a
Range.

The package divides into eight parts. The first part is concerned with the standard constructions
for pre-crossed modules and crossed modules; together with direct products; normal sub-crossed mod-
ules; and quotients. Operations for constructing pre-cat1-groups and cat1-groups, and for converting
between cat1-groups and crossed modules, are also included.

The second part is concerned with morphisms of (pre-)crossed modules and (pre-)cat1-groups,
together with standard operations for morphisms, such as composition, image and kernel.

The third part is the most recent part of the package, introduced in October 2015. Additional
operations and properties for crossed modules are included in Section 4.1. Then, in 4.2 and 4.3 there
are functions for isoclinism of groups and crossed modules.

The fourth part is concerned with the equivalent notions of derivation for a crossed module and
section for a cat1-group, and the monoids which they form under the Whitehead multiplication.

The fifth part deals with actor crossed modules and actor cat1-groups. For the actor crossed mod-
ule Act(X ) of a crossed module X we require representations for the Whitehead group of regular
derivations of X and for the group of automorphisms of X . The construction also provides an inner
morphism from X to Act(X ) whose kernel is the centre of X .

The sixth part, which remains under development, contains functions to compute induced crossed
modules.

Since version 2.007 there are experimental functions for crossed squares and their morphisms,
structures which arise as 3-dimensional groups. Examples of these are inclusions of normal sub-
crossed modules, and the inner morphism from a crossed module to its actor.

The eighth part has some experimental functions for crossed modules of groupoids, interacting
with the package groupoids. Much more work on this is needed.

Future plans include the implementation of group-graphs which will provide examples of pre-
crossed modules (their implementation will require interaction with graph-theoretic functions in GAP
4). There are also plans to implement cat2-groups, and conversion betwen these and crossed squares.

The equivalent categories XMod (crossed modules) and Cat1 (cat1-groups) are also equivalent to
GpGpd, the subcategory of group objects in the category Gpd of groupoids. Finite groupoids have been
implemented in Emma Moore’s package groupoids [Moo01] for groupoids and crossed resolutions.

In order that the user has some control of the verbosity of the XMod package’s functions, an
InfoClass InfoXMod is provided (see Chapter ref:Info Functions in the GAP Reference Manual
for a description of the Info mechanism). By default, the InfoLevel of InfoXMod is 0; progressively
more information is supplied by raising the InfoLevel to 1, 2 and 3.

Example

gap> SetInfoLevel( InfoXMod, 1); #sets the InfoXMod level to 1

https://github.com/gap-packages.github.io/xmod/
https://www.gap-system.org/Packages/packages.html
https://github.com/gap-packages/xmod/

Once the package is loaded, the manual doc/manual.pdf can be found in the documentation
folder. The html versions, with or without MathJax, should be rebuilt as follows:

Example

gap> ReadPackage( "xmod, "makedoc.g" );

It is possible to check that the package has been installed correctly by running the test files:
Example

gap> ReadPackage( "xmod", "tst/testall.g" );
#I Testing .../pkg/xmod/tst/gp2obj.tst
...

Additional information can be found on the Computational Higher-dimensional Discrete Algebra
website at: http://pages.bangor.ac.uk/~mas023/chda/intro.html.

http://pages.bangor.ac.uk/~mas023/chda/intro.html


Chapter 2

2d-groups : crossed modules and
cat1-groups

The term 2d-group refers to a set of equivalent categories of which the most common are the categories
of crossed modules; cat1-groups; and group-groupoids, all of which involve a pair of groups.

2.1 Constructions for crossed modules

A crossed module (of groups) X = (∂ : S→ R) consists of a group homomorphism ∂ , called the
boundary of X , with source S and range R. The group R acts on itself by conjugation, and on S by an
action α : R→ Aut(S) such that, for all s,s1,s2 ∈ S and r ∈ R,

XMod 1 : ∂ (sr) = r−1(∂ s)r = (∂ s)r, XMod 2 : s∂ s2
1 = s−1

2 s1s2 = s1
s2 .

When only the first of these axioms is satisfied, the resulting structure is a pre-crossed module (see
section 2.3). (Much of the literature on crossed modules uses left actions, but we have chosen to use
right actions in this package since that is the standard choice for group actions in GAP.)

The kernel of ∂ is abelian.
There are a variety of constructors for crossed modules:

2.1.1 XMod

. XMod(args) (function)

. XModByBoundaryAndAction(bdy, act) (operation)

. XModByTrivialAction(bdy) (operation)

. XModByNormalSubgroup(G, N) (operation)

. XModByCentralExtension(bdy) (operation)

. XModByAutomorphismGroup(grp) (operation)

. XModByInnerAutomorphismGroup(grp) (operation)

. XModByGroupOfAutomorphisms(G, A) (operation)

. XModByAbelianModule(abmod) (operation)

. DirectProductOp(L, X1) (operation)

The global function XMod implements one of the following standard constructions:

8

• A trivial action crossed module (∂ : S→ R) has sr = s for all s ∈ S, r ∈ R, the source is abelian
and the image lies in the centre of the range.

• A conjugation crossed module is the inclusion of a normal subgroup SER, where R acts on S
by conjugation.

• A central extension crossed module has as boundary a surjection ∂ : S→ R, with central kernel,
where r ∈ R acts on S by conjugation with ∂−1r.

• An automorphism crossed module has as range a subgroup R of the automorphism group Aut(S)
of S which contains the inner automorphism group of S. The boundary maps s ∈ S to the inner
automorphism of S by s.

• A crossed abelian module has an abelian module as source and the zero map as boundary.

• The direct product X1×X2 of two crossed modules has source S1× S2, range R1×R2 and
boundary ∂1×∂2, with R1, R2 acting trivially on S2, S1 respectively.

Since DirectProduct is a global function which only accepts groups, it is necessary to provide
an "other method" for operation DirectProductOp which, as usual, takes as parameters a list
of crossed modules, followed by the first of these: DirectProductOp([X1,X2],X1);

2.1.2 Source (for crossed modules)

. Source(X0) (attribute)

. Range(X0) (attribute)

. Boundary(X0) (attribute)

. XModAction(X0) (attribute)

The following attributes are used in the construction of a crossed module X0.

• Source(X0) and Range(X0) are the source S and range R of ∂ , the boundary Boundary(X0);

• XModAction(X0) is a homomorphism from R to a group of automorphisms of X0.

(Up until version 2.63 there was an additional attribute AutoGroup, the range of XModAction(X0).)

2.1.3 ImageElmXModAction

. ImageElmXModAction(X0, s, r) (operation)

This function returns the element sr given by XModAction(X0).

2.1.4 Size (for crossed modules)

. Size(X0) (attribute)

. Name(X0) (attribute)

. IdGroup(X0) (attribute)

. ExternalSetXMod(X0) (attribute)

More familiar attributes are Name, Size and IdGroup. The name is formed by concatenating the
names of the source and range (if these exist). Size and IdGroup return two-element lists.

The ExternalSetXMod for a crossed module is the source group considered as a G-set of the
range group using the crossed module action.

The Display function is used to print details of 2d-groups.
In the simple example below, X1 is an automorphism crossed module, using a cyclic group of size

five. The Print statements at the end list the GAP representations, properties and attributes of X1.
Example

gap> c5 := Group( (5,6,7,8,9) );;
gap> SetName( c5, "c5" );
gap> X1 := XModByAutomorphismGroup( c5 );
[c5 -> Aut(c5)]
gap> Display( X1 );
Crossed module [c5->Aut(c5)] :-
: Source group c5 has generators:

[ (5,6,7,8,9) ]
: Range group Aut(c5) has generators:

[ GroupHomomorphismByImages( c5, c5, [ (5,6,7,8,9) ], [ (5,7,9,6,8) ] ) ]
: Boundary homomorphism maps source generators to:

[ IdentityMapping( c5 ) ]
: Action homomorphism maps range generators to automorphisms:

GroupHomomorphismByImages( c5, c5, [ (5,6,7,8,9) ],
[ (5,7,9,6,8) ] ) --> { source gens --> [ (5,7,9,6,8) ] }

This automorphism generates the group of automorphisms.
gap> Size( X1 ); IdGroup( X1 );
[ 5, 4 ]
[ [ 5, 1 ], [ 4, 1 ] ]
gap> ext := ExternalSetXMod( X1 );
<xset:[ (), (5,6,7,8,9), (5,7,9,6,8), (5,8,6,9,7), (5,9,8,7,6) ]>
gap> Orbits( ext );
[ [ () ], [ (5,6,7,8,9), (5,7,9,6,8), (5,9,8,7,6), (5,8,6,9,7) ] ]
gap> a := GeneratorsOfGroup( Range( X1 ) )[1]^2;
[ (5,6,7,8,9) ] -> [ (5,9,8,7,6) ]
gap> ImageElmXModAction( X1, (5,7,9,6,8), a );
(5,8,6,9,7)
gap> RepresentationsOfObject( X1 );
[ "IsComponentObjectRep", "IsAttributeStoringRep", "IsPreXModObj" ]
gap> KnownAttributesOfObject( X1);
[ "Name", "Size", "Range", "Source", "IdGroup", "Boundary", "XModAction",

"ExternalSetXMod", "IsomorphismPerm2DimensionalGroup" ]

2.2 Properties of crossed modules

The underlying category structures for the objects constructed in this chapter follow the se-
quence Is2DimensionalDomain; Is2DimensionalMagma; Is2DimensionalMagmaWithOne;
Is2DimensionalMagmaWithInverses, mirroring the situation for (one-dimensional) groups.
From these we construct Is2DimensionalSemigroup, Is2DimensionalMonoid and
Is2DimensionalGroup.

There are then a variety of properties associated with crossed modules, starting with IsPreXMod
and IsXMod.

2.2.1 IsXMod

. IsXMod(X0) (property)

. IsPreXMod(X0) (property)

. IsPerm2DimensionalGroup(X0) (property)

. IsPc2DimensionalGroup(X0) (property)

. IsFp2DimensionalGroup(X0) (property)

A structure which has IsPerm2DimensionalGroup is a precrossed module or a pre-cat1-
group (see section 2.4) whose source and range are both permutation groups. The prop-
erties IsPc2DimensionalGroup, IsFp2DimensionalGroup are defined similarly. In the
example below we see that X1 has IsPreXMod, IsXMod and IsPerm2DimensionalGroup.
There are also properties corresponding to the various construction methods listed in sec-
tion 2.1: IsTrivialAction2DimensionalGroup; IsNormalSubgroup2DimensionalGroup;
IsCentralExtension2DimensionalGroup; IsAutomorphismGroup2DimensionalGroup;
IsAbelianModule2DimensionalGroup.

Example

gap> KnownPropertiesOfObject( X1 );
[ "IsEmpty", "IsTrivial", "IsNonTrivial", "IsFinite",

"CanEasilyCompareElements", "CanEasilySortElements", "IsDuplicateFree",
"IsGeneratorsOfSemigroup", "IsPreXModDomain", "IsPerm2DimensionalGroup",
"IsPreXMod", "IsXMod", "IsAutomorphismGroup2DimensionalGroup" ]

2.2.2 SubXMod

. SubXMod(X0, src, rng) (operation)

. TrivialSubXMod(X0) (attribute)

. NormalSubXMods(X0) (attribute)

With the standard crossed module constructors listed above as building blocks, sub-crossed mod-
ules, normal sub-crossed modules N CX , and also quotients X /N may be constructed. A sub-
crossed module S = (δ : N→M) is normal in X = (∂ : S→ R) if

• N,M are normal subgroups of S,R respectively,

• δ is the restriction of ∂ ,

• nr ∈ N for all n ∈ N, r ∈ R,

• (s−1)ms ∈ N for all m ∈M, s ∈ S.

These conditions ensure that M nN is normal in the semidirect product Rn S. (Note that 〈s,m〉 =
(s−1)ms is a displacement: see Displacement (4.1.3).)

A method for IsNormal for precrossed modules is provided. See section 4.1 for factor crossed
modules and their natural morphisms.

The five normal subcrossed modules of X4 found in the following example are [id,id],
[k4,k4], [k4,a4], [a4,a4] and X4 itself.

Example

gap> s4 := Group( (1,2), (2,3), (3,4) );;
gap> a4 := Subgroup( s4, [ (1,2,3), (2,3,4) ] );;
gap> k4 := Subgroup( a4, [ (1,2)(3,4), (1,3)(2,4) ] );;
gap> SetName(s4,"s4"); SetName(a4,"a4"); SetName(k4,"k4");
gap> X4 := XModByNormalSubgroup( s4, a4 );
[a4->s4]
gap> Y4 := SubXMod( X4, k4, a4 );
[k4->a4]
gap> IsNormal(X4,Y4);
true
gap> NX4 := NormalSubXMods( X4 );;
gap> Length( NX4 );
5

2.3 Pre-crossed modules

2.3.1 PreXModByBoundaryAndAction

. PreXModByBoundaryAndAction(bdy, act) (operation)

. SubPreXMod(X0, src, rng) (operation)

If axiom XMod 2 is not satisfied, the corresponding structure is known as a pre-crossed module.
Example

gap> b1 := (11,12,13,14,15,16,17,18);; b2 := (12,18)(13,17)(14,16);;
gap> d16 := Group( b1, b2 );;
gap> sk4 := Subgroup( d16, [ b1^4, b2 ] );;
gap> SetName( d16, "d16" ); SetName( sk4, "sk4" );
gap> bdy16 := GroupHomomorphismByImages( d16, sk4, [b1,b2], [b1^4,b2] );;
gap> aut1 := GroupHomomorphismByImages( d16, d16, [b1,b2], [b1^5,b2] );;
gap> aut2 := GroupHomomorphismByImages( d16, d16, [b1,b2], [b1,b2^4*b2] );;
gap> aut16 := Group( [ aut1, aut2 ] );;
gap> act16 := GroupHomomorphismByImages( sk4, aut16, [b1^4,b2], [aut1,aut2] );;
gap> P16 := PreXModByBoundaryAndAction( bdy16, act16 );
[d16->sk4]
gap> IsXMod(P16);
false

2.3.2 PeifferSubgroup

. PeifferSubgroup(X0) (attribute)

. XModByPeifferQuotient(prexmod) (attribute)

The Peiffer subgroup P of a pre-crossed module X is the subgroup of ker(∂ ) generated by Peiffer
commutators

bs1,s2c = (s−1
1 )∂ s2 s−1

2 s1 s2 = 〈∂ s2,s1〉 [s1,s2] .

Then P = (0 : P→ {1R}) is a normal sub-pre-crossed module of X and X /P = (∂ : S/P→ R) is
a crossed module.

In the following example the Peiffer subgroup is cyclic of size 4.
Example

gap> P := PeifferSubgroup( P16 );
Group( [ (11,15)(12,16)(13,17)(14,18), (11,17,15,13)(12,18,16,14) ] )
gap> X16 := XModByPeifferQuotient( P16 );
Peiffer([d16->sk4])
gap> Display( X16 );
Crossed module Peiffer([d16->sk4]) :-
: Source group has generators:

[ f1, f2 ]
: Range group has generators:

[ (11,15)(12,16)(13,17)(14,18), (12,18)(13,17)(14,16) ]
: Boundary homomorphism maps source generators to:

[ (12,18)(13,17)(14,16), (11,15)(12,16)(13,17)(14,18) ]
The automorphism group is trivial

gap> iso16 := IsomorphismPermGroup( Source( X16 ) );;
gap> S16 := Image( iso16 );
Group([ (1,2), (3,4) ])

2.4 Cat1-groups and pre-cat1-groups

In [Lod82], Loday reformulated the notion of a crossed module as a cat1-group, namely a group G
with a pair of endomorphisms t,h : G→ G having a common image R and satisfying certain axioms.
We find it computationally convenient to define a cat1-group C = (e; t,h : G→ R) as having source
group G, range group R, and three homomorphisms: two surjections t,h : G→ R and an embedding
e : R→ G satisfying:

Cat 1 : t ◦ e = h◦ e = idR, Cat 2 : [ker t,kerh] = {1G}.

It follows that t ◦ e ◦ h = h, h ◦ e ◦ t = t, t ◦ e ◦ t = t and h ◦ e ◦ h = h. (See section 2.5 for the case
when t,h are endomorphisms.)

2.4.1 Cat1Group

. Cat1Group(args) (function)

. PreCat1Group(args) (function)

. PreCat1GroupByTailHeadEmbedding(t, h, e) (operation)

. PreCat1GroupByEndomorphisms(t, h) (operation)

The global functions Cat1Group and PreCat1Group can be called in various ways.

• as Cat1Group(t,h,e); when t,h,e are three homomorphisms, which is equivalent to
PreCat1GroupByTailHeadEmbedding(t,h,e);

• as Cat1Group(t,h); when t,h are two endomorphisms, which is equivalent to
PreCat1GroupByEndomorphisms(t,h);

• as Cat1Group(t); when t = h is an endomorphism, which is equivalent to
PreCat1GroupByEndomorphisms(t,t);

• as Cat1Group(t,e); when t = h and e are homomorphisms, which is equivalent to
PreCat1GroupByTailHeadEmbedding(t,t,e);

• as Cat1Group(i,j,k); when i, j,k are integers, which is equivalent to Cat1Select(i,j,k);
as described in section 2.6.

Example

gap> g18gens := [ (1,2,3), (4,5,6), (2,3)(5,6) ];;
gap> s3agens := [ (7,8,9), (8,9) ];;
gap> g18 := Group( g18gens );; SetName( g18, "g18" );
gap> s3a := Group( s3agens );; SetName( s3a, "s3a" );
gap> t1 := GroupHomomorphismByImages(g18,s3a,g18gens,[(7,8,9),(),(8,9)]);
[ (1,2,3), (4,5,6), (2,3)(5,6) ] -> [ (7,8,9), (), (8,9) ]
gap> h1 := GroupHomomorphismByImages(g18,s3a,g18gens,[(7,8,9),(7,8,9),(8,9)]);
[ (1,2,3), (4,5,6), (2,3)(5,6) ] -> [ (7,8,9), (7,8,9), (8,9) ]
gap> e1 := GroupHomomorphismByImages(s3a,g18,s3agens,[(1,2,3),(2,3)(5,6)]);
[ (7,8,9), (8,9) ] -> [ (1,2,3), (2,3)(5,6) ]
gap> C18 := Cat1Group( t1, h1, e1 );
[g18=>s3a]

2.4.2 Source (for cat1-groups)

. Source(C) (attribute)

. Range(C) (attribute)

. TailMap(C) (attribute)

. HeadMap(C) (attribute)

. RangeEmbedding(C) (attribute)

. KernelEmbedding(C) (attribute)

. Boundary(C) (attribute)

. Name(C) (attribute)

. Size(C) (attribute)

These are the attributes of a cat1-group C in this implementation.
The maps t,h are often referred to as the source and target, but we choose to call them the tail and

head of C , because source is the GAP term for the domain of a function. The RangeEmbedding is the
embedding of R in G, the KernelEmbedding is the inclusion of the kernel of t in G, and the Boundary
is the restriction of h to the kernel of t. It is frequently the case that t = h, but not in the example C18
above.

Example

gap> Source( C18 );
g18
gap> Range( C18 );
s3a
gap> TailMap( C18 );
[ (1,2,3), (4,5,6), (2,3)(5,6) ] -> [ (7,8,9), (), (8,9) ]
gap> HeadMap( C18 );
[ (1,2,3), (4,5,6), (2,3)(5,6) ] -> [ (7,8,9), (7,8,9), (8,9) ]
gap> RangeEmbedding( C18 );
[ (7,8,9), (8,9) ] -> [ (1,2,3), (2,3)(5,6) ]
gap> Kernel( C18 );
Group([ (4,5,6) ])
gap> KernelEmbedding( C18 );
[ (4,5,6) ] -> [ (4,5,6) ]
gap> Name( C18 );
"[g18=>s3a]"
gap> Size( C18 );
[ 18, 6 ]
gap> StructureDescription( C18 );
[ "(C3 x C3) : C2", "S3" ]

2.4.3 DiagonalCat1Group

. DiagonalCat1Group(gen1) (operation)

. PreCat1GroupByNormalSubgroup(G, N) (operation)

. Cat1GroupByPeifferQuotient(P) (operation)

. ReverseCat1Group(C0) (attribute)

These are some more constructors for cat1-groups. The following listing shows an ex-
ample of a permutation cat1-group of size [576,24] with source group S4 × S4, range group
a third S4, and t 6= h. A similar example may be reproduced using the command C :=
DiagonalCat1Group([(1,2,3,4),(3,4)]);.

Example

gap> G4 := Group( (1,2,3,4), (3,4), (5,6,7,8), (7,8) );;
gap> R4 := Group( (9,10,11,12), (11,12) );;
gap> SetName( G4, "s4s4" ); SetName( R4, "s4d" );
gap> G4gens := GeneratorsOfGroup( G4 );;
gap> R4gens := GeneratorsOfGroup( R4 );;
gap> t := GroupHomomorphismByImages( G4, R4, G4gens,
> Concatenation( R4gens, [ (), () ] ) );;
gap> h := GroupHomomorphismByImages( G4, R4, G4gens,
> Concatenation( [ (), () ], R4gens ) );;
gap> e := GroupHomomorphismByImages( R4, G4, R4gens,
> [ (1,2,3,4)(5,6,7,8), (3,4)(7,8) ] );;
gap> C4 := PreCat1GroupByTailHeadEmbedding( t, h, e );;
gap> Display(C4);
Cat1-group [s4s4=>s4d] :-

: Source group s4s4 has generators:
[ (1,2,3,4), (3,4), (5,6,7,8), (7,8) ]

: Range group s4d has generators:
[ ( 9,10,11,12), (11,12) ]

: tail homomorphism maps source generators to:
[ ( 9,10,11,12), (11,12), (), () ]

: head homomorphism maps source generators to:
[ (), (), ( 9,10,11,12), (11,12) ]

: range embedding maps range generators to:
[ (1,2,3,4)(5,6,7,8), (3,4)(7,8) ]

: kernel has generators:
[ (5,6,7,8), (7,8) ]

: boundary homomorphism maps generators of kernel to:
[ ( 9,10,11,12), (11,12) ]

: kernel embedding maps generators of kernel to:
[ (5,6,7,8), (7,8) ]

2.5 Properties of cat1-groups and pre-cat1-groups

Many of the properties listed in section 2.2 apply to pre-cat1-groups and to cat1-groups since these
are also 2d-groups. There are also more specific properties.

2.5.1 IsCat1Group

. IsCat1Group(C0) (property)

. IsPreXCat1Group(C0) (property)

. IsIdentityCat1Group(C0) (property)

. IsEndomorphismPreCat1Group(C0) (property)

. EndomorphismPreCat1Group(C0) (attribute)

IsIdentityCat1Group(C0) is true when the head and tail maps of C0 are identity mappings.
IsEndomorphismPreCat1Group(C0) is true when the range of C0 is a subgroup of the source. When
this is not the case, replacing t,h,e by t ∗ e,h∗ e and the inclusion mapping of the image of e gives an
isomorphic cat1-group for which IsEndomorphismPreCat1Group is true.

Example

gap> G2 := SmallGroup( 288, 956 ); SetName( G2, "G2" );
<pc group of size 288 with 7 generators>
gap> d12 := DihedralGroup( 12 ); SetName( d12, "d12" );
<pc group of size 12 with 3 generators>
gap> a1 := d12.1;; a2 := d12.2;; a3 := d12.3;; a0 := One( d12 );;
gap> gensG2 := GeneratorsOfGroup( G2 );;
gap> t2 := GroupHomomorphismByImages( G2, d12, gensG2,
> [ a0, a1*a3, a2*a3, a0, a0, a3, a0 ] );;
gap> h2 := GroupHomomorphismByImages( G2, d12, gensG2,
> [ a1*a2*a3, a0, a0, a2*a3, a0, a0, a3^2 ] );;
gap> e2 := GroupHomomorphismByImages( d12, G2, [a1,a2,a3],
> [ G2.1*G2.2*G2.4*G2.6^2, G2.3*G2.4*G2.6^2*G2.7, G2.6*G2.7^2 ] );

[ f1, f2, f3 ] -> [ f1*f2*f4*f6^2, f3*f4*f6^2*f7, f6*f7^2 ]
gap> C2 := PreCat1GroupByTailHeadEmbedding( t2, h2, e2 );
[G2=>d12]
gap> IsCat1Group( C2 );
true
gap> KnownPropertiesOfObject( C2 );
[ "CanEasilyCompareElements", "CanEasilySortElements", "IsDuplicateFree",

"IsGeneratorsOfSemigroup", "IsPreCat1Domain", "IsPerm2DimensionalGroup",
"IsPreCat1Group", "IsCat1Group", "IsEndomorphismPreCat1Group" ]

gap> IsEndomorphismPreCat1Group( C2 );
false
gap> EC4 := EndomorphismPreCat1Group( C4 );
[s4s4=>Group( [ (1,2,3,4)(5,6,7,8), (3,4)(7,8), (), () ] )]

2.5.2 Cat1GroupOfXMod

. Cat1GroupOfXMod(X0) (attribute)

. XModOfCat1Group(C0) (attribute)

. PreCat1GroupOfPreXMod(P0) (attribute)

. PreXModOfPreCat1Group(P0) (attribute)

The category of crossed modules is equivalent to the category of cat1-groups, and the functors
between these two categories may be described as follows. Starting with the crossed module X =
(∂ : S→ R) the group G is defined as the semidirect product G = RnS using the action from X , with
multiplication rule

(r1,s1)(r2,s2) = (r1r2,s1
r2s2).

The structural morphisms are given by

t(r,s) = r, h(r,s) = r(∂ s), er = (r,1).

On the other hand, starting with a cat1-group C = (e; t,h : G→ R), we define S = ker t, the range R is
unchanged, and ∂ = h |S. The action of R on S is conjugation in G via the embedding of R in G.

Example

gap> X2 := XModOfCat1Group( C2 );;
gap> Display( X2 );

Crossed module X([G2=>d12]) :-
: Source group has generators:

[ f1, f4, f5, f7 ]
: Range group d12 has generators:

[ f1, f2, f3 ]
: Boundary homomorphism maps source generators to:

[ f1*f2*f3, f2*f3, <identity> of ..., f3^2 ]
: Action homomorphism maps range generators to automorphisms:

f1 --> { source gens --> [ f1*f5, f4*f5, f5, f7^2 ] }
f2 --> { source gens --> [ f1*f5*f7^2, f4, f5, f7 ] }
f3 --> { source gens --> [ f1*f7, f4, f5, f7 ] }
These 3 automorphisms generate the group of automorphisms.

: associated cat1-group is [G2=>d12]

gap> StructureDescription(X2);
[ "D24", "D12" ]

2.6 Selection of a small cat1-group

The Cat1Group function may also be used to select a cat1-group from a data file. All cat1-
structures on groups of size up to 70 (ordered according to the GAP 4 numbering of small
groups) are stored in a list in file cat1data.g. Global variables CAT1_LIST_MAX_SIZE := 70 and
CAT1_LIST_CLASS_SIZES are also stored. The data is read into the list CAT1_LIST only when this
function is called.

2.6.1 Cat1Select

. Cat1Select(size, gpnum, num) (operation)

The function Cat1Select may be used in three ways. Cat1Select( size ) returns the names
of the groups with this size, while Cat1Select( size, gpnum ) prints a list of cat1-structures for
this chosen group. Cat1Select( size, gpnum, num ) returns the chosen cat1-group.

The example below is the first case in which t 6= h and the associated conjugation crossed module
is given by the normal subgroup c3 of s3.

Example

gap> ## check the number of groups of size 18
gap> L18 := Cat1Select( 18 );
Usage: Cat1Select( size, gpnum, num );
[ "D18", "C18", "C3 x S3", "(C3 x C3) : C2", "C6 x C3" ]
gap> ## check the number of cat1-structures on the fourth of these
gap> Cat1Select( 18, 4 );
Usage: Cat1Select( size, gpnum, num );
There are 4 cat1-structures for the group (C3 x C3) : C2.
Using small generating set [ f1, f2, f2*f3 ] for source of homs.
[ [range gens], [tail genimages], [head genimages] ] :-
(1) [ [ f1 ], [ f1, <identity> of ..., <identity> of ... ],

[ f1, <identity> of ..., <identity> of ... ] ]
(2) [ [ f1, f3 ], [ f1, <identity> of ..., f3 ],

[ f1, <identity> of ..., f3 ] ]
(3) [ [ f1, f3 ], [ f1, <identity> of ..., f3 ],

[ f1, f3^2, <identity> of ... ] ]
(4) [ [ f1, f2, f2*f3 ], tail = head = identity mapping ]
4
gap> ## select the third of these cat1-structures
gap> C18 := Cat1Group( 18, 4, 3 );
[(C3 x C3) : C2=>Group( [ f1, <identity> of ..., f3 ] )]
gap> ## convert from a pc-cat1-group to a permutation cat1-group
gap> iso18 := IsomorphismPermObject( C18 );;



XMod 19

gap> PC18 := Image( iso18 );;
gap> Display( PC18 );
Cat1-group :-
: Source group has generators:

[ (2,3)(5,6), (4,5,6), (1,2,3) ]
: Range group has generators:

[ (2,3), (), (1,2,3) ]
: tail homomorphism maps source generators to:

[ (2,3), (), (1,2,3) ]
: head homomorphism maps source generators to:

[ (2,3), (1,3,2), (1,2,3) ]
: range embedding maps range generators to:

[ (2,3)(5,6), (), (1,2,3) ]
: kernel has generators:

[ (4,5,6) ]
: boundary homomorphism maps generators of kernel to:

[ (1,3,2) ]
: kernel embedding maps generators of kernel to:

[ (4,5,6) ]
gap> convert the result to the associated permutation crossed module
gap> X18 := XModOfCat1Group( PC18 );;
gap> Display( X18 );
Crossed module:-
: Source group has generators:

[ (4,5,6) ]
: Range group has generators:

[ (2,3), (), (1,2,3) ]
: Boundary homomorphism maps source generators to:

[ (1,3,2) ]
: Action homomorphism maps range generators to automorphisms:

(2,3) --> { source gens --> [ (4,6,5) ] }
() --> { source gens --> [ (4,5,6) ] }
(1,2,3) --> { source gens --> [ (4,5,6) ] }
These 3 automorphisms generate the group of automorphisms.

: associated cat1-group is [..=>..]

2.6.2 AllCat1DataGroupsBasic

. AllCat1DataGroupsBasic(gp) (operation)

For a group G of size greater than 70 which is reasonably straightforward this function may be
used to construct a list of all cat1-group structures on G. The operation also attempts to write output
to a file in the folder xmod/lib. (Other operations in the file cat1data.gi have been used to deal
with the more complicated groups of size up to 70, but these are not described here.)

Van Luyen Le has a more efficient algorithm, extending the data up to groups of size 171, which
is expected to appear in a future release of HAP.

Example

gap> gp := SmallGroup( 102, 2 );
<pc group of size 102 with 3 generators>



XMod 20

gap> StructureDescription( gp );
"C3 x D34"
gap> all := AllCat1DataGroupsBasic( gp );
#I Edit last line of .../xmod/lib/nn.kk.out to end with ] ] ] ] ]
[ [Group( [ f1, f2, f3 ] )=>Group( [ f1, <identity> of ..., <identity> of ...

] )], [Group( [ f1, f2, f3 ] )=>Group( [ f1, f2, <identity> of ... ] )],
[Group( [ f1, f2, f3 ] )=>Group( [ f1, <identity> of ..., f3 ] )],
[Group( [ f1, f2, f3 ] )=>Group( [ f1, f2, f3 ] )] ]

2.7 More functions for crossed modules and cat1-groups

Chapter 4 contains functions for quotient crossed modules; centre of a crossed module; commutator
and derived subcrossed modules; etc.

Here we mention two functions for groups which have been extended to the two-dimensional case.

2.7.1 IdGroup (for 2d-groups)

. IdGroup(2DimensionalGroup) (operation)

. StructureDescription(2DimensionalGroup) (operation)

These functions return two-element lists formed by applying the function to the source and range
of the 2d-group.

Example

gap> IdGroup( X2 );
[ [ 24, 6 ], [ 12, 4 ] ]
gap> StructureDescription( C2 );
[ "(S3 x D24) : C2", "D12" ]



Chapter 3

2d-mappings

3.1 Morphisms of 2-dimensional groups

This chapter describes morphisms of (pre-)crossed modules and (pre-)cat1-groups.

3.1.1 Source (for 2d-group mappings)

. Source(map) (attribute)

. Range(map) (attribute)

. SourceHom(map) (attribute)

. RangeHom(map) (attribute)

Morphisms of 2-dimensional groups are implemented as 2-dimensional mappings. These have a
pair of 2-dimensional groups as source and range, together with two group homomorphisms mapping
between corresponding source and range groups. These functions return fail when invalid data is
supplied.

3.2 Morphisms of pre-crossed modules

3.2.1 IsXModMorphism

. IsXModMorphism(map) (property)

. IsPreXModMorphism(map) (property)

A morphism between two pre-crossed modules X1 = (∂1 : S1 → R1) and X2 = (∂2 : S2 → R2)
is a pair (σ ,ρ), where σ : S1 → S2 and ρ : R1 → R2 commute with the two boundary maps and are
morphisms for the two actions:

∂2 ◦σ = ρ ◦∂1, σ(sr) = (σs)ρr.

Here σ is the SourceHom (3.1.1) and ρ is the RangeHom (3.1.1) of the morphism. When X1 = X2
and σ ,ρ are automorphisms then (σ ,ρ) is an automorphism of X1. The group of automorphisms is
denoted by Aut(X1).

21



XMod 22

3.2.2 IsInjective (for pre-xmod morphisms)

. IsInjective(map) (method)

. IsSurjective(map) (method)

. IsSingleValued(map) (method)

. IsTotal(map) (method)

. IsBijective(map) (method)

. IsEndo2DimensionalMapping(map) (property)

The usual properties of mappings are easily checked. It is usually sufficient to verify that both the
SourceHom (3.1.1) and the RangeHom (3.1.1) have the required property.

3.2.3 XModMorphism

. XModMorphism(args) (function)

. XModMorphismByHoms(X1, X2, sigma, rho) (operation)

. PreXModMorphism(args) (function)

. PreXModMorphismByHoms(P1, P2, sigma, rho) (operation)

. InclusionMorphism2DimensionalDomains(X1, S1) (operation)

. InnerAutomorphismXMod(X1, r) (operation)

. IdentityMapping(X1) (attribute)

These are the constructors for morphisms of pre-crossed and crossed modules.
In the following example we construct a simple automorphism of the crossed module X1 con-

structed in the previous chapter.
Example

gap> sigma1 := GroupHomomorphismByImages( c5, c5, [ (5,6,7,8,9) ]
[ (5,9,8,7,6) ] );;

gap> rho1 := IdentityMapping( Range( X1 ) );
IdentityMapping( PAut(c5) )
gap> mor1 := XModMorphism( X1, X1, sigma1, rho1 );
[[c5->Aut(c5))] => [c5->Aut(c5))]]
gap> Display( mor1 );
Morphism of crossed modules :-
: Source = [c5->Aut(c5)] with generating sets:

[ (5,6,7,8,9) ]
[ GroupHomomorphismByImages( c5, c5, [ (5,6,7,8,9) ], [ (5,7,9,6,8) ] ) ]

: Range = Source
: Source Homomorphism maps source generators to:

[ (5,9,8,7,6) ]
: Range Homomorphism maps range generators to:

[ GroupHomomorphismByImages( c5, c5, [ (5,6,7,8,9) ], [ (5,7,9,6,8) ] ) ]
gap> IsAutomorphism2DimensionalDomain( mor1 );
true
gap> Order( mor1 );
2
gap> RepresentationsOfObject( mor1 );
[ "IsComponentObjectRep", "IsAttributeStoringRep", "Is2DimensionalMappingRep" ]
gap> KnownPropertiesOfObject( mor1 );

[ "CanEasilyCompareElements", "CanEasilySortElements", "IsTotal",
"IsSingleValued", "IsInjective", "IsSurjective", "RespectsMultiplication",
"IsPreXModMorphism", "IsXModMorphism", "IsEndomorphism2DimensionalDomain",
"IsAutomorphism2DimensionalDomain" ]

gap> KnownAttributesOfObject( mor1 );
[ "Name", "Order", "Range", "Source", "SourceHom", "RangeHom" ]

3.2.4 IsomorphismPerm2DimensionalGroup (for pre-xmod morphisms)

. IsomorphismPerm2DimensionalGroup(obj) (attribute)

. IsomorphismPc2DimensionalGroup(obj) (attribute)

. IsomorphismByIsomorphisms(D, list) (operation)

When D is a 2-dimensional domain with source S and range R and σ : S → S′, ρ : R → R′

are isomorphisms, then IsomorphismByIsomorphisms(D,[sigma,rho]) returns an isomorphism
(σ ,ρ) : D → D ′ where D ′ has source S′ and range R′. Be sure to test IsBijective for the two
functions σ ,ρ before applying this operation.

Using IsomorphismByIsomorphisms with a pair of isomorphisms obtained using
IsomorphismPermGroup or IsomorphismPcGroup, we may construct a crossed module or a
cat1-group of permutation groups or pc-groups.

Example

gap> q8 := SmallGroup(8,4);; ## quaternion group
gap> Xq8 := XModByAutomorphismGroup( q8 );
[Group( [ f1, f2, f3 ] )->Group( [ f1, f2, f3, f4 ] )]
gap> iso := IsomorphismPerm2DimensionalGroup( Xq8 );;
gap> Yq8 := Image( iso );
[Group( [ (1,2,4,6)(3,8,7,5), (1,3,4,7)(2,5,6,8), (1,4)(2,6)(3,7)(5,8)
] )->Group( [ (2,6,5,4), (1,2,4)(3,5,6), (2,5)(4,6), (1,3)(2,5) ] )]

gap> s4 := SymmetricGroup(4);;
gap> isos4 := IsomorphismGroups( Range(Yq8), s4 );;
gap> id := IdentityMapping( Source( Yq8 ) );;
gap> IsBijective( id );; IsBijective( isos4 );;
gap> mor := IsomorphismByIsomorphisms( Yq8, [id,isos4] );;
gap> Zq8 := Image( mor );
[Group( [ (1,2,4,6)(3,8,7,5), (1,3,4,7)(2,5,6,8), (1,4)(2,6)(3,7)(5,8)
] )->SymmetricGroup( [ 1 .. 4 ] )]

3.3 Morphisms of pre-cat1-groups

A morphism of pre-cat1-groups from C1 = (e1; t1,h1 : G1→ R1) to C2 = (e2; t2,h2 : G2→ R2) is a pair
(γ,ρ) where γ : G1→ G2 and ρ : R1→ R2 are homomorphisms satisfying

h2 ◦ γ = ρ ◦h1, t2 ◦ γ = ρ ◦ t1, e2 ◦ρ = γ ◦ e1.

3.3.1 IsCat1Morphism

. IsCat1Morphism(map) (property)

. IsPreCat1Morphism(map) (property)

. Cat1Morphism(args) (function)

. Cat1MorphismByHoms(C1, C2, gamma, rho) (operation)

. PreCat1Morphism(args) (function)

. PreCat1MorphismByHoms(P1, P2, gamma, rho) (operation)

. InclusionMorphism2DimensionalDomains(C1, S1) (operation)

. InnerAutomorphismCat1(C1, r) (operation)

. IdentityMapping(C1) (attribute)

For an example we form a second cat1-group C2=[g18=>s3a], similar to C1 in 2.4.1, then con-
struct an isomorphism (γ,ρ) between them.

Example

gap> t2 := GroupHomomorphismByImages(g18,s3a,g18gens,[(),(7,8,9),(8,9)]);;
gap> e2 := GroupHomomorphismByImages(s3a,g18,s3agens,[(4,5,6),(2,3)(5,6)]);;
gap> C2 := Cat1Group( t2, h1, e2 );;
gap> imgamma := [ (4,5,6), (1,2,3), (2,3)(5,6) ];;
gap> gamma := GroupHomomorphismByImages( g18, g18, g18gens, imgamma );;
gap> rho := IdentityMapping( s3a );;
gap> mor := Cat1Morphism( C1, C2, gamma, rho );;
gap> Display( mor );;
Morphism of cat1-groups :-
: Source = [g18=>s3a] with generating sets:

[ (1,2,3), (4,5,6), (2,3)(5,6) ]
[ (7,8,9), (8,9) ]

: Range = [g18=>s3a] with generating sets:
[ (1,2,3), (4,5,6), (2,3)(5,6) ]
[ (7,8,9), (8,9) ]

: Source Homomorphism maps source generators to:
[ (4,5,6), (1,2,3), (2,3)(5,6) ]

: Range Homomorphism maps range generators to:
[ (7,8,9), (8,9) ]

3.3.2 IsomorphismPermObject

. IsomorphismPermObject(obj) (function)

. IsomorphismPerm2DimensionalGroup(2DimensionalGroup) (attribute)

. IsomorphismFp2DimensionalGroup(2DimensionalGroup) (attribute)

. IsomorphismPc2DimensionalGroup(2DimensionalGroup) (attribute)

. SmallerDegreePerm2DimensionalDomain(2DimensionalDomain) (function)

The global function IsomorphismPermObject calls IsomorphismPerm2DimensionalGroup,
which constructs a morphism whose SourceHom (3.1.1) and RangeHom (3.1.1) are calcu-
lated using IsomorphismPermGroup on the source and range. Similarly the operation
SmallerDegreePermutationRepresentation may be used on the two groups to obtain the attribute

SmallerDegreePerm2DimensionalDomain. Names are assigned automatically.
Example

gap> iso2 := IsomorphismPerm2DimensionalGroup( C2 );
[[G2=>d12] => [..]]
gap> Display( iso2 );
Morphism of cat1-groups :-
: Source = [G2=>d12] with generating sets:

[ f1, f2, f3, f4, f5, f6, f7 ]
[ f1, f2, f3 ]

: Range = P[G2=>d12] with generating sets:
[ ( 6,12)( 8,15)( 9,16)(11,19)(13,26)(14,22)(17,27)(18,25)(20,21)(23,24),
( 2, 3)( 5,10)( 9,16)(11,18)(17,23)(19,25)(24,27),
( 4, 5, 7,10)( 6, 9,12,16)( 8,11,14,18)(13,17,20,23)(15,19,22,25)

(21,24,26,27), ( 4, 6, 7,12)( 5, 9,10,16)( 8,13,14,20)(11,17,18,23)
(15,21,22,26)(19,24,25,27), ( 4, 7)( 5,10)( 6,12)( 8,14)( 9,16)(11,18)
(13,20)(15,22)(17,23)(19,25)(21,26)(24,27), ( 1, 2, 3),

( 4, 8,15)( 5,11,19)( 6,13,21)( 7,14,22)( 9,17,24)(10,18,25)(12,20,26)
(16,23,27) ]

[ (2,6)(3,5), (1,2,3,4,5,6), (1,3,5)(2,4,6) ]
: Source Homomorphism maps source generators to:

[ ( 6,12)( 8,15)( 9,16)(11,19)(13,26)(14,22)(17,27)(18,25)(20,21)(23,24),
( 2, 3)( 5,10)( 9,16)(11,18)(17,23)(19,25)(24,27),
( 4, 5, 7,10)( 6, 9,12,16)( 8,11,14,18)(13,17,20,23)(15,19,22,25)

(21,24,26,27), ( 4, 6, 7,12)( 5, 9,10,16)( 8,13,14,20)(11,17,18,23)
(15,21,22,26)(19,24,25,27), ( 4, 7)( 5,10)( 6,12)( 8,14)( 9,16)(11,18)
(13,20)(15,22)(17,23)(19,25)(21,26)(24,27), ( 1, 2, 3),

( 4, 8,15)( 5,11,19)( 6,13,21)( 7,14,22)( 9,17,24)(10,18,25)(12,20,26)
(16,23,27) ]

: Range Homomorphism maps range generators to:
[ (2,6)(3,5), (1,2,3,4,5,6), (1,3,5)(2,4,6) ]

3.4 Operations on morphisms

3.4.1 CompositionMorphism

. CompositionMorphism(map2, map1) (operation)

Composition of morphisms (written (<map1> * <map2>) when maps act on the right) calls the
CompositionMorphism function for maps (acting on the left), applied to the appropriate type of 2d-
mapping.

Example

gap> H2 := Subgroup(G2,[G2.3,G2.4,G2.6,G2.7]); SetName( H2, "H2" );
Group([ f3, f4, f6, f7 ])
gap> c6 := Subgroup( d12, [b,c] ); SetName( c6, "c6" );
Group([ f2, f3 ])
gap> SC2 := Sub2DimensionalGroup( C2, H2, c6 );
[H2=>c6]
gap> IsCat1Group( SC2 );

true
gap> inc2 := InclusionMorphism2DimensionalDomains( C2, SC2 );
[[H2=>c6] => [G2=>d12]]
gap> CompositionMorphism( iso2, inc );
[[H2=>c6] => P[G2=>d12]]

3.4.2 Kernel (for 2d-mappings)

. Kernel(map) (operation)

. Kernel2DimensionalMapping(map) (attribute)

The kernel of a morphism of crossed modules is a normal subcrossed module whose groups are
the kernels of the source and target homomorphisms. The inclusion of the kernel is a standard example
of a crossed square, but these have not yet been implemented.

Example

gap> c2 := Group( (19,20) );
Group([ (19,20) ])
gap> X0 := XModByNormalSubgroup( c2, c2 ); SetName( X0, "X0" );
[Group( [ (19,20) ] )->Group( [ (19,20) ] )]
gap> SX2 := Source( X2 );;
gap> genSX2 := GeneratorsOfGroup( SX2 );
[ f1, f4, f5, f7 ]
gap> sigma0 := GroupHomomorphismByImages(SX2,c2,genSX2,[(19,20),(),(),()]);
[ f1, f4, f5, f7 ] -> [ (19,20), (), (), () ]
gap> rho0 := GroupHomomorphismByImages(d12,c2,[a1,a2,a3],[(19,20),(),()]);
[ f1, f2, f3 ] -> [ (19,20), (), () ]
gap> mor0 := XModMorphism( X2, X0, sigma0, rho0 );;
gap> K0 := Kernel( mor0 );;
gap> StructureDescription( K0 );
[ "C12", "C6" ]



Chapter 4

Isoclinism of groups and crossed modules

This chapter describes some functions written by Alper Odabaş and Enver Uslu, and reported in their
paper [IOU16]. Section 4.1 contains some additional basic functions for crossed modules, constructing
quotients, centres, centralizers and normalizers. In Sections 4.2 and 4.3 there are functions dealing
specifically with isoclinism for groups and for crossed modules. Since these functions represent a
recent addition to the package (as of November 2015), the function names are liable to change in
future versions. The notion of isoclinism has been crucial to the enumeration of groups of prime
power order, see for example James, Newman and O’Brien, [JNO90].

4.1 More operations for crossed modules

4.1.1 FactorPreXMod

. FactorPreXMod(X1, X2) (operation)

. NaturalMorphismByNormalSubPreXMod(X1, X2) (operation)

When X2 = (∂2 : S2 → R2) is a normal sub-precrossed module of X1 = (∂1 : S1 → R1), then
the quotient precrossed module is (∂ : S2/S1→ R2/R1) with the induced boundary and action maps.
Quotienting a precrossed module by it’s Peiffer subgroup is a special case of this construction.

Example

gap> d24 := DihedralGroup( IsPermGroup, 24 );;
gap> SetName( d24, "d24" );
gap> Y24 := XModByAutomorphismGroup( d24 );;
gap> Size( Y24 );
[ 24, 48 ]
gap> X24 := Image( IsomorphismPerm2DimensionalGroup( Y24 ) );
[d24->Group( [ ( 2, 4), ( 1, 2, 3, 4)( 5, 8)( 6, 9)( 7,10), ( 6,10)( 7, 9),

( 5, 9, 7)( 6,10, 8) ] )]
gap> nsx := NormalSubXMods( X24 );;
gap> Length( nsx );
40
gap> ids := List( nsx, n -> IdGroup(n) );;
gap> pos1 := Position( ids, [ [4,1], [8,3] ] );;
gap> Xn1 := nsx[pos1];
[Group( [ f2*f4^2, f3*f4 ] )->Group( [ f3, f4, f5 ] )]

27

gap> nat1 := NaturalMorphismByNormalSubPreXMod( X24, Xn1 );;
gap> Qn1 := FactorPreXMod( X24, Xn1 );;
gap> [ Size( Xn1 ), Size( Qn1 ) ];
[ [ 4, 8 ], [ 6, 6 ] ]

4.1.2 IntersectionSubXMods

. IntersectionSubXMods(X0, X1, X2) (operation)

When X1,X2 are subcrossed modules of X0, then the source and range of their intersection are the
intersections of the sources and ranges of X1 and X2 respectively.

Example

gap> pos2 := Position( ids, [ [24,6], [12,4] ] );;
gap> Xn2 := nsx[pos2];;
gap> IdGroup( Xn2 );
[ [ 24, 6 ], [ 12, 4 ] ]
gap> pos3 := Position( ids, [ [12,2], [24,5] ] );;
gap> Xn3 := nsx[pos3];;
gap> IdGroup( Xn3 );
[ [ 12, 2 ], [ 24, 5 ] ]
gap> Xn23 := IntersectionSubXMods( X24, Xn2, Xn3 );;
gap> IdGroup( Xn23 );
[ [ 12, 2 ], [ 6, 2 ] ]

4.1.3 Displacement

. Displacement(alpha, r, s) (operation)

. DisplacementSubgroup(X0) (attribute)

Commutators may be written [r,q] = r−1q−1rq = (q−1)rq = r−1rq, and satisfy identities

[r,q]p = [rp,qp], [pr,q] = [p,q]r[r,q], [r, pq] = [r,q][r, p]q, [r,q]−1 = [q,r].

In a similar way, when a group R acts on a group S, the displacement of s ∈ S by r ∈ R is defined to be
〈r,s〉 := (s−1)rs ∈ S. When X = (∂ : S→ R) is a pre-crossed module, the first crossed module axiom
requires ∂ 〈r,s〉 = [r,∂ s]. For a given action α the Displacement function may be used to calculate
〈r,s〉. Displacements satisfy the following identities, where s, t ∈ S, p,q,r ∈ R:

〈r,s〉p = 〈rp,sp〉, 〈qr,s〉= 〈q,s〉r〈r,s〉, 〈r,st〉= 〈r, t〉〈r,s〉t , 〈r,s〉−1 = 〈r−1,sr〉.

The DisplacementSubgroup of X is the subgroup Disp(X ) of S generated by these displacements.
The identities imply 〈r,s〉t = 〈r,str−1〉〈r−1, t〉, so Disp(X ) is normal in S.

Example

gap> pos4 := Position( ids, [ [6,2], [24,14] ] );;
gap> Xn4 := nsx[pos4];;

gap> bn4 := Boundary( Xn4 );;
gap> Sn4 := Source(Xn4);;
gap> Rn4 := Range(Xn4);;
gap> genRn4 := GeneratorsOfGroup( Rn4 );;
gap> L := List( genRn4, g -> ( Order(g) = 2 ) and
> not ( IsNormal( Rn4, Subgroup( Rn4, [g] ) ) ) );;
gap> pos := Position( L, true );;
gap> s := Sn4.1; r := genRn4[pos];
(1,3,5,7,9,11)(2,4,6,8,10,12)
(6,10)(7,9)
gap> act := XModAction( Xn4 );;
gap> d := Displacement( act, r, s );
(1,5,9)(2,6,10)(3,7,11)(4,8,12)
gap> Image( bn4, d ) = Comm( r, Image( bn4, s ) );
true
gap> DisplacementSubgroup( Xn4 );
Group([ (1,5,9)(2,6,10)(3,7,11)(4,8,12) ])

4.1.4 CommutatorSubXMod

. CommutatorSubXMod(X, X1, X2) (operation)

. CrossActionSubgroup(X, X1, X2) (operation)

When X1 =(N→Q),X2 =(M→P) are two normal subcrossed modules of X =(∂ : S→R), the
displacements 〈p,n〉 and 〈q,m〉 all map by ∂ into [Q,P]. These displacements form a normal subgroup
of S, called the CrossActionSubgroup. The CommutatorSubXMod [X1,X2] has this subgroup as
source and [P,Q] as range, and is normal in X .

Example

gap> CAn23 := CrossActionSubgroup( X24, Xn2, Xn3 );;
gap> IdGroup( CAn23 );
[ 12, 2 ]
gap> Cn23 := CommutatorSubXMod( X24, Xn2, Xn3 );;
gap> IdGroup( Cn23 );
[ [ 12, 2 ], [ 6, 2 ] ]
gap> Xn23 = Cn23;
true

4.1.5 DerivedSubXMod

. DerivedSubXMod(X0) (attribute)

The DerivedSubXMod of X is the normal subcrossed module [X ,X ] = (∂ ′ : Disp(X )→ [R,R])
where ∂ ′ is the restriction of ∂ (see page 66 of Norrie’s thesis [Nor87]).

Example

gap> DXn4 := DerivedSubXMod( Xn4 );;
gap> IdGroup( DXn4 );

[ [ 3, 1 ], [ 3, 1 ] ]

4.1.6 FixedPointSubgroupXMod

. FixedPointSubgroupXMod(X0, T, Q) (operation)

. StabilizerSubgroupXMod(X0, T, Q) (operation)

The FixedPointSubgroupXMod(X,T,Q) for X = (∂ : S→ R) is the subgroup Fix(X ,T,Q) of
elements t ∈ T 6 S individually fixed under the action of Q6 R.

The StabilizerSubgroupXMod(X,T,Q) for X is the subgroup Stab(X ,T,Q) of Q6 R whose
elements act trivially on the whole of T 6 S (see page 19 of Norrie’s thesis [Nor87]).

Example

gap> fix := FixedPointSubgroupXMod( Xn4, Sn4, Rn4 );
Group([ (1,7)(2,8)(3,9)(4,10)(5,11)(6,12) ])
gap> stab := StabilizerSubgroupXMod( Xn4, Sn4, Rn4 );;
gap> IdGroup( stab );
[ 12, 5 ]

4.1.7 CentreXMod

. CentreXMod(X0) (attribute)

. Centralizer(X, Y) (operation)

. Normalizer(X, Y) (operation)

The centre Z(X ) of X = (∂ : S→ R) has as source the fixed point subgroup Fix(X ,S,R). The
range is the intersection of the centre Z(R) with the stabilizer subgroup.

When Y = (T → Q) is a subcrossed module of X = (∂ : S→ R), the centralizer CX (Y ) of Y
has as source the fixed point subgroup Fix(X ,S,Q). The range is the intersection of the centralizer
CR(Q) with Stab(X ,T,R).

The normalizer NX (Y ) of Y has as source the subgroup of S consisting of the displacements
〈s,q〉 which lie in S.

Example

gap> ZXn4 := CentreXMod( Xn4 );
[Group( [ f3*f4 ] )->Group( [ f3, f5 ] )]
gap> IdGroup( ZXn4 );
[ [ 2, 1 ], [ 4, 2 ] ]
gap> CDXn4 := Centralizer( Xn4, DXn4 );
[Group( [ f3*f4 ] )->Group( [ f2 ] )]
gap> IdGroup( CDXn4 );
[ [ 2, 1 ], [ 3, 1 ] ]
gap> NDXn4 := Normalizer( Xn4, DXn4 );
[Group( <identity> of ... )->Group( [ f5, f2*f3 ] )]
gap> IdGroup( NDXn4 );
[ [ 1, 1 ], [ 12, 5 ] ]

4.1.8 CentralQuotient

. CentralQuotient(G) (attribute)

The CentralQuotient of a group G is the crossed module (G→ G/Z(G)) with the natural ho-
momorphism as the boundary map. This is a special case of XModByCentralExtension (2.1.1).

Similarly, the central quotient of a crossed module X is the crossed square (X ⇒X /Z(X ))
(see section 8.2).

Example

gap> Q24 := CentralQuotient( d24); IdGroup( Q24 );
[d24->d24/Z(d24)]
[ [ 24, 6 ], [ 12, 4 ] ]

4.1.9 IsAbelian2DimensionalGroup

. IsAbelian2DimensionalGroup(X0) (property)

. IsAspherical2DimensionalGroup(X0) (property)

. IsSimplyConnected2DimensionalGroup(X0) (property)

. IsFaithful2DimensionalGroup(X0) (property)

A crossed module is abelian if it equal to its centre. This is the case when the range group is
abelian and the action is trivial.

A crossed module is aspherical if the boundary has trivial kernel.
A crossed module is simply connected if the boundary has trivial cokernel.
A crossed module is faithful if the action is faithful.

Example

gap> [ IsAbelian2DimensionalGroup(Xn4), IsAbelian2DimensionalGroup(X24) ];
[ false, false ]
gap> pos7 := Position( ids, [ [3,1], [6,1] ] );;
gap> [ IsAspherical2DimensionalGroup(nsx[pos7]), IsAspherical2DimensionalGroup(X24) ];
[ true, false ]
gap> [ IsSimplyConnected2DimensionalGroup(Xn4), IsSimplyConnected2DimensionalGroup(X24) ];
[ true, true ]
gap> [ IsFaithful2DimensionalGroup(Xn4), IsFaithful2DimensionalGroup(X24) ];
[ false, true ]

4.1.10 LowerCentralSeriesOfXMod

. LowerCentralSeriesOfXMod(X0) (attribute)

. IsNilpotent2DimensionalGroup(X0) (property)

. NilpotencyClass2DimensionalGroup(X0) (attribute)

Let Y be a subcrossed module of X . A series of length n from X to Y has the form

X = X0 D X1 D · · · D Xi D · · · D Xn = Y (16 i6 n).

The quotients Fi = Xi/Xi−1 are the factors of the series.
A factor Fi is central if Xi−1EX and Fi is a subcrossed module of the centre of X /Xi−1.
A series is central if all its factors are central.
X is soluble if it has a series all of whose factors are abelian.
X is nilpotent is it has a series all of whose factors are central factors of X .
The lower central series of X is the sequence (see [Nor87], p.77):

X = Γ1(X ) D Γ2(X ) D · · · where Γ j(X ) = [Γ j−1(X ),X ].

If X is nilpotent, then its lower central series is its most rapidly descending central series.
The least integer c such that Γc+1(X ) is the trivial crossed module is the nilpotency class of X .

Example

gap> lcs := LowerCentralSeries( X24 );;
gap> List( lcs, g -> IdGroup(g) );
[ [ [ 24, 6 ], [ 48, 38 ] ], [ [ 12, 2 ], [ 6, 2 ] ], [ [ 6, 2 ], [ 3, 1 ] ],

[ [ 3, 1 ], [ 3, 1 ] ] ]
gap> IsNilpotent2DimensionalGroup( X24 );
false
gap> NilpotencyClassOf2DimensionalGroup( X24 );
0

4.1.11 AllXMods

. AllXMods(args) (function)

The global function AllXMods may be called in three ways: as AllXMods(S,R) to compute all
crossed modules with chosen source and range groups; as AllXMods([n,m]) to compute all crossed
modules with a given size; or as AllXMods(ord) to compute all crossed modules whose associated
cat1-groups have a given size ord.

In the example we see that there are 4 crossed modules (C6 → S3); forming a subset of the 17
crossed modules with size [6,6]; and that these form a subset of the 205 crossed modules whose
cat1-group has size 36. There are 40 precrossed modules with size [6,6].

Example

gap> xc6s3 := AllXMods( SmallGroup(6,2), SmallGroup(6,1) );;
gap> Length( xc6s3 );
4
gap> x66 := AllXMods( [6,6] );;
gap> Length( x66 );
17
gap> x36 := AllXMods( 36 );;
gap> Length( x36 );
205
gap> size36 := List( x36, x -> [ Size(Source(x)), Size(Range(x)) ] );;
gap> Collected( size36 );
[ [ [ 1, 36 ], 14 ], [ [ 2, 18 ], 7 ], [ [ 3, 12 ], 21 ], [ [ 4, 9 ], 14 ],

[ [ 6, 6 ], 17 ], [ [ 9, 4 ], 102 ], [ [ 12, 3 ], 8 ], [ [ 18, 2 ], 18 ],
[ [ 36, 1 ], 4 ] ]

4.1.12 IsomorphismXMods

. IsomorphismXMods(X1, X2) (operation)

. AllXModsUpToIsomorphism(list) (operation)

The function IsomorphismXMods computes an isomorphism µ : X1→X2, provided one exists,
or else returns fail. In the example below we see that the 17 crossed modules of size [6,6] in x66
(see the previous subsection) fall into 9 isomorphism classes.

The function AllXModsUpToIsomorphism takes a list of crossed modules and partitions them into
isomorphism classes.

Example

gap> IsomorphismXMods( x66[1], x66[2] );
[[Group( [ f1, f2 ] )->Group( [ f1, f2 ] )] => [Group( [ f1, f2 ] )->Group(
[ f1, f2 ] )]]
gap> iso66 := AllXModsUpToIsomorphism( x66 );; Length( iso66 );
9

4.2 Isoclinism for groups

4.2.1 Isoclinism (for groups)

. Isoclinism(G, H) (operation)

. AreIsoclinicDomains(G, H) (operation)

Let G,H be groups with central quotients Q(G) and Q(H) and derived subgroups [G,G] and [H,H]
respectively. Let cG : G/Z(G)×G/Z(G)→ [G,G] and cH : H/Z(H)×H/Z(H)→ [H,H] be the two
commutator maps. An isoclinism G ∼ H is a pair of isomorphisms (η ,ξ ) where η : G/Z(G)→
H/Z(H) and ξ : [G,G]→ [H,H] such that cG ∗ξ = (η×η)∗cH . Isoclinism is an equivalence relation,
and all abelian groups are isoclinic to the trivial group.

Example

gap> G := SmallGroup( 64, 6 );; StructureDescription( G );
"(C8 x C4) : C2"
gap> QG := CentralQuotient( G );; IdGroup( QG );
[ [ 64, 6 ], [ 8, 3 ] ]
gap> H := SmallGroup( 32, 41 );; StructureDescription( H );
"C2 x Q16"
gap> QH := CentralQuotient( H );; IdGroup( QH );
[ [ 32, 41 ], [ 8, 3 ] ]
gap> Isoclinism( G, H );
[ [ f1, f2, f3 ] -> [ f1, f2*f3, f3 ], [ f3, f5 ] -> [ f4*f5, f5 ] ]
gap> K := SmallGroup( 32, 43 );; StructureDescription( K );
"(C2 x D8) : C2"
gap> QK := CentralQuotient( K );; IdGroup( QK );
[ [ 32, 43 ], [ 16, 11 ] ]
gap> AreIsoclinicDomains( G, K );
false

4.2.2 IsStemDomain (for groups)

. IsStemDomain(G) (property)

. IsoclinicStemDomain(G) (attribute)

. AllStemGroupIds(n) (operation)

. AllStemGroupFamilies(n) (operation)

A group G is a stem group if Z(G)≤ [G,G]. Every group is isoclinic to a stem group, but distinct
stem groups may be isoclinic. For example, groups D8,Q8 are two isoclinic stem groups.

The function IsoclinicStemDomain returns a stem group isoclinic to G.
The function AllStemGroupIds returns the IdGroup list of the stem groups of a specified size,

while AllStemGroupFamilies splits this list into isoclinism classes.
Example

gap> DerivedSubgroup(G);
Group([ f3, f5 ])
gap> IsStemDomain( G );
false
gap> IsoclinicStemDomain( G );
<pc group of size 16 with 4 generators>
gap> AllStemGroupIds( 32 );
[ [ 32, 6 ], [ 32, 7 ], [ 32, 8 ], [ 32, 18 ], [ 32, 19 ], [ 32, 20 ],

[ 32, 27 ], [ 32, 28 ], [ 32, 29 ], [ 32, 30 ], [ 32, 31 ], [ 32, 32 ],
[ 32, 33 ], [ 32, 34 ], [ 32, 35 ], [ 32, 43 ], [ 32, 44 ], [ 32, 49 ],
[ 32, 50 ] ]

gap> AllStemGroupFamilies( 32 );
[ [ [ 32, 6 ], [ 32, 7 ], [ 32, 8 ] ], [ [ 32, 18 ], [ 32, 19 ], [ 32, 20 ] ],

[ [ 32, 27 ], [ 32, 28 ], [ 32, 29 ], [ 32, 30 ], [ 32, 31 ], [ 32, 32 ],
[ 32, 33 ], [ 32, 34 ], [ 32, 35 ] ], [ [ 32, 43 ], [ 32, 44 ] ],

[ [ 32, 49 ], [ 32, 50 ] ] ]

4.2.3 IsoclinicRank (for groups)

. IsoclinicRank(G) (attribute)

. IsoclinicMiddleLength(G) (attribute)

Let G be a finite p-group. Then logp |[G,G]/(Z(G)∩ [G,G])| is called the middle length of G.
Also logp |Z(G)∩ [G,G]|+ logp |G/Z(G)| is called the rank of G. These invariants appear in the tables
of isoclinism families of groups of order 128 in [JNO90].

Example

gap> IsoclinicMiddleLength( G );
1
gap> IsoclinicRank( G );
4

4.3 Isoclinism for crossed modules

4.3.1 Isoclinism (for crossed modules)

. Isoclinism(X0, Y0) (operation)

. AreIsoclinicDomains(X0, Y0) (operation)

Let X ,Y be crossed modules with central quotients Q(X ) and Q(Y ), and derived subcrossed
modules [X ,X ] and [Y ,Y ] respectively. Let cX : Q(X )×Q(X )→ [X ,X ] and cY : Q(Y )×
Q(Y )→ [Y ,Y ] be the two commutator maps. An isoclinism X ∼Y is a pair of bijective morphisms
(η ,ξ ) where η : Q(X )→ Q(Y ) and ξ : [X ,X ]→ [Y ,Y ] such that cX ∗ ξ = (η ×η) ∗ cY . Iso-
clinism is an equivalence relation, and all abelian crossed modules are isoclinic to the trivial crossed
module.

Example

gap> C8 := Cat1Group( 16, 8, 1 );;
gap> X8 := XMod(C8); IdGroup( X8 );
[Group( [ f1*f2*f3, f3, f4 ] )->Group( [ f2, f2 ] )]
[ [ 8, 1 ], [ 2, 1 ] ]
gap> C9 := Cat1Group( 32, 9, 1 );
[(C8 x C2) : C2=>Group( [ f2, f2 ] )]
gap> X9 := XMod( C9 ); IdGroup( X9 );
[Group( [ f1*f2*f3, f3, f4, f5 ] )->Group( [ f2, f2 ] )]
[ [ 16, 5 ], [ 2, 1 ] ]
gap> AreIsoclinicDomains( X8, X9 );
true
gap> ism89 := Isoclinism( X8, X9 );;
gap> Display( ism89 );
[ [[Group( [ f1, f2, <identity> of ... ] ) -> Group( [ f2, f2 ] )] => [Group(

[ f1, f2, <identity> of ..., <identity> of ... ] ) -> Group(
[ f2, f2 ] )]],

[[Group( [ f3 ] ) -> Group( <identity> of ... )] => [Group(
[ f3 ] ) -> Group( <identity> of ... )]] ]

4.3.2 IsStemDomain (for crossed modules of groups)

. IsStemDomain(X0) (property)

. IsoclinicStemDomain(X0) (attribute)

A crossed module X is a stem crossed module if Z(X ) ≤ [X ,X ]. Every crossed module is
isoclinic to a stem crossed module, but distinct stem crossed modules may be isoclinic.

A method for IsoclinicStemDomain has yet to be implemented.
Example

gap> IsStemDomain(X8);
true
gap> IsStemDomain(X9);
false

4.3.3 IsoclinicRank (for crossed modules of groups)

. IsoclinicRank(X0) (attribute)

. IsoclinicMiddleLength(X0) (attribute)

The formulae in subsection 4.2.3 are applied to the crossed module.
Example

gap> IsoclinicMiddleLength(X8);
[ 1, 0 ]
gap> IsoclinicRank(X8);
[ 3, 1 ]



Chapter 5

Whitehead group of a crossed module

5.1 Derivations and Sections

The Whitehead monoid Der(X ) of X was defined in [Whi48] to be the monoid of all derivations
from R to S, that is the set of all maps χ : R→ S, with Whitehead multiplication ? (on the right)
satisfying:

Der 1 : χ(qr) = (χq)r (χr), Der 2 : (χ1 ?χ2)(r) = (χ2r)(χ1r)(χ2∂ χ1r).

The zero map is the identity for this composition. Invertible elements in the monoid are called regular.
The Whitehead group of X is the group of regular derivations in Der(X ). In the next chapter the
actor of X is defined as a crossed module whose source and range are permutation representations of
the Whitehead group and the automorphism group of X .

The construction for cat1-groups equivalent to the derivation of a crossed module is the section.
The monoid of sections of C = (e; t,h : G→ R) is the set of group homomorphisms ξ : R→ G, with
Whitehead multiplication ? (on the right) satisfying:

Sect 1 : t ◦ξ = idR, Sect 2 : (ξ1 ?ξ2)(r) = (ξ1r)(ehξ1r)−1(ξ2hξ1r) = (ξ2hξ1r)(ehξ1r)−1(ξ1r).

The embedding e is the identity for this composition, and h(ξ1 ?ξ2) = (hξ1)(hξ2). A section is regular
when hξ is an automorphism, and the group of regular sections is isomorphic to the Whitehead group.

If ε denotes the inclusion of S = ker t in G then ∂ = hε : S→ R and

ξ r = (er)(eχr), which equals (r,χr) ∈ RnS,

determines a section ξ of C in terms of the corresponding derivation χ of X , and conversely.

5.1.1 DerivationByImages

. DerivationByImages(X0, ims) (operation)

. IsDerivation(map) (property)

. IsUp2DimensionalMapping(map) (property)

. UpImagePositions(chi) (attribute)

. UpGeneratorImages(chi) (attribute)

Derivations are stored like group homomorphisms by specifying the images of a generating
set. Images of the remaining elements may then be obtained using axiom Der 1. The function
IsDerivation is automatically called to check that this procedure is well-defined.

In the following example a cat1-group C3 and the associated crossed module X3 are constructed,
where X3 is isomorphic to the inclusion of the normal cyclic group c3 in the symmetric group s3.

Example

gap> g18 := Group( (1,2,3), (4,5,6), (2,3)(5,6) );;
gap> SetName( g18, "g18" );
gap> gen18 := GeneratorsOfGroup( g18 );;
gap> g1 := gen18[1];; g2 := gen18[2];; g3 := gen18[3];;
gap> s3 := Subgroup( g18, gen18{[2..3]} );;
gap> SetName( s3, "s3" );;
gap> t := GroupHomomorphismByImages( g18, s3, gen18, [g2,g2,g3] );;
gap> h := GroupHomomorphismByImages( g18, s3, gen18, [(),g2,g3] );;
gap> e := GroupHomomorphismByImages( s3, g18, [g2,g3], [g2,g3] );;
gap> C3 := Cat1Group( t, h, e );
[g18=>s3]
gap> SetName( Kernel(t), "c3" );;
gap> X3 := XModOfCat1Group( C3 );
[c3->s3]
gap> imchi1 := [ (), (1,2,3)(4,6,5) ];;
gap> chi1 := DerivationByImages( X3, imchi1 );
DerivationByImages( s3, c3, [ (4,5,6), (2,3)(5,6) ],
[ (), (1,2,3)(4,6,5) ] )
gap> [ IsUp2DimensionalMapping( chi1 ), IsDerivation( chi1 ) ];
[ true, true ]
gap> UpImagePositions( chi1 );
[ 1, 1, 1, 2, 2, 2 ]
gap> UpGeneratorImages( chi1 );
[ (), (1,2,3)(4,6,5) ]

5.1.2 PrincipalDerivation

. PrincipalDerivation(X0, s) (operation)

The principal derivation determined by s ∈ S is the derivation ηs : R→ S, r 7→ (s−1)rs.
Example

gap> eta := PrincipalDerivation( X3, (1,2,3)(4,6,5) );
DerivationByImages( s3, c3, [ (4,5,6), (2,3)(5,6) ], [ (), (1,3,2)(4,5,6) ] )

5.1.3 SectionByHomomorphism

. SectionByHomomorphism(C, hom) (operation)

. IsSection(hom) (property)

. SectionByDerivation(chi) (operation)

. DerivationBySection(xi) (operation)

Sections are group homomorphisms, so do not need a special representation. Operations
SectionByDerivation and DerivationBySection convert derivations to sections, and vice-versa,
calling Cat1GroupOfXMod (2.5.2) and XModOfCat1Group (2.5.2) automatically.

Two strategies for calculating derivations and sections are implemented, see [AW00]. The default
method for AllDerivations (5.2.1) is to search for all possible sets of images using a backtracking
procedure, and when all the derivations are found it is not known which are regular. In early versions
of this package, the default method for AllSections( <C> ) was to compute all endomorphisms on
the range group R of C as possibilities for the composite hξ . A backtrack method then found possible
images for such a section. In the current version the derivations of the associated crossed module are
calculated, and these are all converted to sections using SectionByDerivation.

Example

gap> hom2 := GroupHomomorphismByImages( s3, g18, [ (4,5,6), (2,3)(5,6) ],
> [ (1,3,2)(4,6,5), (1,2)(4,6) ] );;
gap> xi2 := SectionByHomomorphism( C3, hom2 );
SectionByHomomorphism( s3, g18, [ (4,5,6), (2,3)(5,6) ],
[ (1,3,2)(4,6,5), (1,2)(4,6) ] )
gap> [ IsUp2DimensionalMapping( xi2 ), IsSection( xi2 ) ];
[ true, true ]
gap> chi2 := DerivationBySection( xi2 );
DerivationByImages( s3, c3, [ (4,5,6), (2,3)(5,6) ],
[ (1,3,2)(4,5,6), (1,2,3)(4,6,5) ] )
gap> xi1 := SectionByDerivation( chi1 );
SectionByHomomorphism( s3, g18, [ (4,5,6), (2,3)(5,6) ],
[ (1,2,3), (1,2)(4,6) ] )

5.1.4 IdentityDerivation

. IdentityDerivation(X0) (attribute)

. IdentitySection(C0) (attribute)

The identity derivation maps the range group to the identity subgroup of the source, while the
identity section is just the range embedding considered as a section.

Example

gap> IdentityDerivation( X3 );
DerivationByImages( s3, c3, [ (4,5,6), (2,3)(5,6) ], [ (), () ] )
gap> IdentitySection(C3);
SectionByHomomorphism( s3, g18, [ (4,5,6), (2,3)(5,6) ],
[ (4,5,6), (2,3)(5,6) ] )

5.1.5 WhiteheadProduct

. WhiteheadProduct(chi1, chi2) (operation)

. WhiteheadOrder(chi) (operation)

The WhiteheadProduct may be applied to two derivations to form χ1 ? χ2, or to two sections to
form ξ1 ? ξ2. The WhiteheadOrder of a regular derivation χ is the smallest power of χ , using this
product, equal to the IdentityDerivation (5.1.4).

Example

gap> chi12 := WhiteheadProduct( chi1, chi2 );
DerivationByImages( s3, c3, [ (4,5,6), (2,3)(5,6) ], [ (1,2,3)(4,6,5), () ] )
gap> xi12 := WhiteheadProduct( xi1, xi2 );
SectionByHomomorphism( s3, g18, [ (4,5,6), (2,3)(5,6) ],
[ (1,2,3), (2,3)(5,6) ] )
gap> xi12 = SectionByDerivation( chi12 );
true
gap> [ WhiteheadOrder( chi2 ), WhiteheadOrder( xi2 ) ];
[ 2, 2 ]

5.2 Whitehead Groups and Monoids

As mentioned at the beginning of this chapter, the Whitehead monoid Der(X ) of X is
the monoid of all derivations from R to S. Monoids of derivations have representation
IsMonoidOfUp2DimensionalMappingsObj. Multiplication tables for Whitehead monoids enable
the construction of transformation representations.

5.2.1 AllDerivations

. AllDerivations(X0) (attribute)

. ImagesTable(obj) (attribute)

. DerivationClass(mon) (attribute)

. WhiteheadMonoidTable(X0) (attribute)

. WhiteheadTransformationMonoid(X0) (attribute)

Using our example X3 we find that there are just nine derivations.
Example

gap> all3 := AllDerivations( X3 );
monoid of derivations with images list:
[ [ (), () ],

[ (), (1,3,2)(4,5,6) ],
[ (), (1,2,3)(4,6,5) ],
[ (1,3,2)(4,5,6), () ],
[ (1,3,2)(4,5,6), (1,3,2)(4,5,6) ],
[ (1,3,2)(4,5,6), (1,2,3)(4,6,5) ],
[ (1,2,3)(4,6,5), () ],
[ (1,2,3)(4,6,5), (1,3,2)(4,5,6) ],
[ (1,2,3)(4,6,5), (1,2,3)(4,6,5) ]
]

gap> DerivationClass( all3 );
"all"

gap> PrintOneItemPerLine( ImagesTable( all3 ) );
[ [ 1, 1, 1, 1, 1, 1 ],

[ 1, 1, 1, 3, 3, 3 ],
[ 1, 1, 1, 2, 2, 2 ],
[ 1, 3, 2, 1, 3, 2 ],
[ 1, 3, 2, 3, 2, 1 ],
[ 1, 3, 2, 2, 1, 3 ],
[ 1, 2, 3, 1, 2, 3 ],
[ 1, 2, 3, 3, 1, 2 ],
[ 1, 2, 3, 2, 3, 1 ]
]

gap> wmt3 := WhiteheadMonoidTable( X3 );;
gap> PrintOneItemPerLine( wmt3 );
[ [ 1, 2, 3, 4, 5, 6, 7, 8, 9 ],

[ 2, 3, 1, 5, 6, 4, 8, 9, 7 ],
[ 3, 1, 2, 6, 4, 5, 9, 7, 8 ],
[ 4, 6, 5, 1, 3, 2, 7, 9, 8 ],
[ 5, 4, 6, 2, 1, 3, 8, 7, 9 ],
[ 6, 5, 4, 3, 2, 1, 9, 8, 7 ],
[ 7, 7, 7, 7, 7, 7, 7, 7, 7 ],
[ 8, 8, 8, 8, 8, 8, 8, 8, 8 ],
[ 9, 9, 9, 9, 9, 9, 9, 9, 9 ] ]

gap> wtm3 := WhiteheadTransformationMonoid( X3 );
<transformation monoid of degree 9 with 3 generators>
gap> GeneratorsOfMonoid( wtm3 );
[ Transformation( [ 2, 3, 1, 5, 6, 4, 8, 9, 7 ] ),

Transformation( [ 4, 6, 5, 1, 3, 2, 7, 9, 8 ] ),
Transformation( [ 7, 7, 7, 7, 7, 7, 7, 7, 7 ] ) ]

5.2.2 RegularDerivations

. RegularDerivations(X0) (attribute)

. ImagesList(obj) (attribute)

. WhiteheadGroupTable(X0) (attribute)

. WhiteheadPermGroup(X0) (attribute)

RegularDerivations are those derivations which are invertible in the monoid. Multiplication
tables for the Whitehead group enable the construction of permutation representations.

Of the nine derivations of X3 just six are regular. The associated group is isomorphic to the
symmetric group s3.

Example

gap> reg3 := RegularDerivations( X3 );
monoid of derivations with images list:
[ [ (), () ],

[ (), (1,3,2)(4,5,6) ],
[ (), (1,2,3)(4,6,5) ],
[ (1,3,2)(4,5,6), () ],
[ (1,3,2)(4,5,6), (1,3,2)(4,5,6) ],
[ (1,3,2)(4,5,6), (1,2,3)(4,6,5) ] ]

gap> wgt3 := WhiteheadGroupTable( X3 );;
gap> PrintOneItemPerLine( wgt3 );
[ [ 1, 2, 3, 4, 5, 6 ],

[ 2, 3, 1, 5, 6, 4 ],
[ 3, 1, 2, 6, 4, 5 ],
[ 4, 6, 5, 1, 3, 2 ],
[ 5, 4, 6, 2, 1, 3 ],
[ 6, 5, 4, 3, 2, 1 ] ]

gap> wpg3 := WhiteheadPermGroup( X3 );
Group([ (1,2,3)(4,5,6), (1,4)(2,6)(3,5) ])

5.2.3 PrincipalDerivations

. PrincipalDerivations(X0) (attribute)

The principal derivations form a subgroup of the Whitehead group.
Example

gap> PDX3 := PrincipalDerivations( X3 );
monoid of derivations with images list:
[ [ (), () ],

[ (), (1,3,2)(4,5,6) ],
[ (), (1,2,3)(4,6,5) ]
]

5.2.4 AllSections

. AllSections(C0) (attribute)

. RegularSections(C0) (attribute)

These operations have been declared but are not yet implemented. The interested user should,
instead, work with the corresponding derivations.



Chapter 6

Actors of 2d-groups

6.1 Actor of a crossed module

The actor of X is a crossed module Act(X ) = (∆ : W (X )→ Aut(X )) which was shown by Lue
and Norrie, in [Nor87] and [Nor90] to give the automorphism object of a crossed module X . In this
implementation, the source of the actor is a permutation representation W of the Whitehead group of
regular derivations, and the range of the actor is a permutation representation A of the automorphism
group Aut(X ) of X .

6.1.1 AutomorphismPermGroup

. AutomorphismPermGroup(xmod) (attribute)

. GeneratingAutomorphisms(xmod) (attribute)

. PermAutomorphismAsXModMorphism(xmod, perm) (operation)

The automorphisms (σ ,ρ) of X form a group Aut(X ) of crossed module isomorphisms. The
function AutomorphismPermGroup finds a set of GeneratingAutomorphisms for Aut(X ), and then
constructs a permutation representation of this group, which is used as the range of the actor crossed
module of X . The individual automorphisms can be constructed from the permutation group using
the function PermAutomorphismAsXModMorphism. The example below uses the crossed module
X3=[c3->s3] constructed in section 5.1.

Example

gap> APX3 := AutomorphismPermGroup( X3 );
Group([ (5,7,6), (1,2)(3,4)(6,7) ])
gap> Size( APX3 );
6
gap> genX3 := GeneratingAutomorphisms( X3 );
[ [[c3->s3] => [c3->s3]], [[c3->s3] => [c3->s3]] ]
gap> e6 := Elements( APX3 )[6];
(1,2)(3,4)(5,7)
gap> m6 := PermAutomorphismAsXModMorphism( X3, e6 );;
gap> Display( m6 );
Morphism of crossed modules :-
: Source = [c3->s3] with generating sets:

[ (1,2,3)(4,6,5) ]

[ (4,5,6), (2,3)(5,6) ]
: Range = Source
: Source Homomorphism maps source generators to:

[ (1,3,2)(4,5,6) ]
: Range Homomorphism maps range generators to:

[ (4,6,5), (2,3)(4,5) ]

6.1.2 WhiteheadXMod

. WhiteheadXMod(xmod) (attribute)

. LueXMod(xmod) (attribute)

. NorrieXMod(xmod) (attribute)

. ActorXMod(xmod) (attribute)

An automorphism (σ ,ρ) of X acts on the Whitehead monoid by χ(σ ,ρ) = σ ◦ χ ◦ ρ−1, and this
determines the action for the actor. In fact the four groups R,S,W,A, the homomorphisms between
them, and the various actions, give five crossed modules forming a crossed square:

• X = (∂ : S→ R), the initial crossed module, on the left,

• W (X ) = (η : S→W ), the Whitehead crossed module of X , at the top,

• N (X) = (α : R→ A), the Norrie crossed module of X , at the bottom,

• Act(X ) = (∆ : W → A), the actor crossed module of X , on the right, and

• L (X ) = (∆◦η = α ◦∂ : S→ A), the Lue crossed module of X , along the top-left to bottom-
right diagonal.

Example

gap> WGX3 := WhiteheadPermGroup( X3 );
Group([ (1,2,3)(4,5,6), (1,4)(2,6)(3,5) ])
gap> WX3 := WhiteheadXMod( X3 );;
gap> Display( WX3 );
Crossed module Whitehead[c3->s3] :-
: Source group has generators:

[ (1,2,3)(4,6,5) ]
: Range group has generators:

[ (1,2,3)(4,5,6), (1,4)(2,6)(3,5) ]
: Boundary homomorphism maps source generators to:

[ (1,2,3)(4,5,6) ]
: Action homomorphism maps range generators to automorphisms:

(1,2,3)(4,5,6) --> { source gens --> [ (1,2,3)(4,6,5) ] }
(1,4)(2,6)(3,5) --> { source gens --> [ (1,3,2)(4,5,6) ] }
These 2 automorphisms generate the group of automorphisms.

gap> LX3 := LueXMod( X3 );;
gap> Display( LX3 );
Crossed module Lue[c3->s3] :-
: Source group has generators:

[ (1,2,3)(4,6,5) ]

: Range group has generators:
[ (5,7,6), (1,2)(3,4)(6,7) ]

: Boundary homomorphism maps source generators to:
[ (5,7,6) ]

: Action homomorphism maps range generators to automorphisms:
(5,7,6) --> { source gens --> [ (1,2,3)(4,6,5) ] }
(1,2)(3,4)(6,7) --> { source gens --> [ (1,3,2)(4,5,6) ] }
These 2 automorphisms generate the group of automorphisms.

gap> NX3 := NorrieXMod( X3 );;
gap> Display( NX3 );
Crossed module Norrie[c3->s3] :-
: Source group has generators:

[ (4,5,6), (2,3)(5,6) ]
: Range group has generators:

[ (5,7,6), (1,2)(3,4)(6,7) ]
: Boundary homomorphism maps source generators to:

[ (5,6,7), (1,2)(3,4)(6,7) ]
: Action homomorphism maps range generators to automorphisms:

(5,7,6) --> { source gens --> [ (4,5,6), (2,3)(4,5) ] }
(1,2)(3,4)(6,7) --> { source gens --> [ (4,6,5), (2,3)(5,6) ] }
These 2 automorphisms generate the group of automorphisms.

gap> AX3 := ActorXMod( X3 );;
gap> Display( AX3);
Crossed module Actor[c3->s3] :-
: Source group has generators:

[ (1,2,3)(4,5,6), (1,4)(2,6)(3,5) ]
: Range group has generators:

[ (5,7,6), (1,2)(3,4)(6,7) ]
: Boundary homomorphism maps source generators to:

[ (5,7,6), (1,2)(3,4)(6,7) ]
: Action homomorphism maps range generators to automorphisms:

(5,7,6) --> { source gens --> [ (1,2,3)(4,5,6), (1,6)(2,5)(3,4) ] }
(1,2)(3,4)(6,7) --> { source gens --> [ (1,3,2)(4,6,5), (1,4)(2,6)(3,5) ] }
These 2 automorphisms generate the group of automorphisms.

gap> IAX3 := InnerActorXMod( X3 );;
gap> Display( IAX3 );
Crossed module InnerActor[c3->s3] :-
: Source group has generators:

[ (1,2,3)(4,5,6) ]
: Range group has generators:

[ (5,6,7), (1,2)(3,4)(6,7) ]
: Boundary homomorphism maps source generators to:

[ (5,7,6) ]
: Action homomorphism maps range generators to automorphisms:

(5,6,7) --> { source gens --> [ (1,2,3)(4,5,6) ] }
(1,2)(3,4)(6,7) --> { source gens --> [ (1,3,2)(4,6,5) ] }
These 2 automorphisms generate the group of automorphisms.

6.1.3 XModCentre

. XModCentre(xmod) (attribute)

. InnerActorXMod(xmod) (attribute)

. InnerMorphism(xmod) (attribute)

Pairs of boundaries or identity mappings provide six morphisms of crossed modules. In particular,
the boundaries of W (X ) and N (X ) form the inner morphism of X , mapping source elements
to principal derivations and range elements to inner automorphisms. The image of X under this
morphism is the inner actor of X , while the kernel is the centre of X . In the example which follows,
the inner morphism of X3=(c3->s3), from Chapter 5, is an inclusion of crossed modules.

Note that we appear to have defined two sorts of centre for a crossed module: XModCentre here,
and CentreXMod (4.1.7) in the chapter on isoclinism. We suspect that these two definitions give the
same answer, but this remains to be resolved.

Example

gap> IMX3 := InnerMorphism( X3 );;
gap> Display( IMX3 );
Morphism of crossed modules :-
: Source = [c3->s3] with generating sets:

[ (1,2,3)(4,6,5) ]
[ (4,5,6), (2,3)(5,6) ]

: Range = Actor[c3->s3] with generating sets:
[ (1,2,3)(4,5,6), (1,4)(2,6)(3,5) ]
[ (5,7,6), (1,2)(3,4)(6,7) ]

: Source Homomorphism maps source generators to:
[ (1,2,3)(4,5,6) ]

: Range Homomorphism maps range generators to:
[ (5,6,7), (1,2)(3,4)(6,7) ]

gap> IsInjective( IMX3 );
true
gap> ZX3 := XModCentre( X3 );
[Group( () )->Group( () )]

Induced constructions

Before describing general functions for computing induced structures, we consider coproducts of
crossed modules which provide induced crossed modules in certain cases.

7.1 Coproducts of crossed modules

Need to add here a reference (or two) for coproducts.

7.1.1 CoproductXMod

. CoproductXMod(X1, X2) (operation)

. CoproductInfo(X0) (attribute)

This function calculates the coproduct crossed module of crossed modules X1 = (∂1 : S1 → R)
and X2 = (∂2 : S2→ R) which have a common range R. The source S2 of X2 acts on S1 via ∂2 and the
action of X1, so we can form a precrossed module (∂ ′ : S1 nS2→ R) where ∂ ′(s1,s2) = (∂1s1)(∂2s2).
The action of this precrossed module is the diagonal action (s1,s2)

r = (sr
1,s

r
2). Factoring out by the

Peiffer subgroup, we obtain the coproduct crossed module X1 ◦X2.
In the example the structure descriptions of the precrossed module, the Peiffer subgroup, and the

resulting coproduct are printed out when InfoLevel(InfoXMod} is at least 1. The coproduct comes
supplied with attribute CoproductInfo, which includes the embedding morphisms of the two factors.

Example

gap> q8 := Group( (1,2,3,4)(5,8,7,6), (1,5,3,7)(2,6,4,8) );;
gap> X8 := XModByAutomorphismGroup( q8 );;
gap> s4b := Range( X8 );;
gap> SetName( q8, "q8" ); SetName( s4b, "s4b" );
gap> a := q8.1;; b := q8.2;;
gap> alpha := GroupHomomorphismByImages( q8, q8, [a,b], [a^-1,b] );;
gap> beta := GroupHomomorphismByImages( q8, q8, [a,b], [a,b^-1] );;
gap> k4b := Subgroup( s4b, [ alpha, beta ] );; SetName( k4b, "k4b" );
gap> Z8 := XModByNormalSubgroup( s4b, k4b );;
gap> SetName( X8, "X8" ); SetName( Z8, "Z8" );
gap> SetInfoLevel( InfoXMod, 1 );
gap> XZ8 := CoproductXMod( X8, Z8 );
#I prexmod is [ [ 32, 47 ], [ 24, 12 ] ]

#I peiffer subgroup is C2, [ 2, 1 ]
#I the coproduct is [ "C2 x C2 x C2 x C2", "S4" ], [ [ 16, 14 ], [ 24, 12 ] ]
[Group( [ f1, f2, f3, f4 ] )->s4b]
gap> SetName( XZ8, "XZ8" );
gap> info := CoproductInfo( XZ8 );
rec( embeddings := [ [X8 => XZ8], [Z8 => XZ8] ], xmods := [ X8, Z8 ] )
gap> SetInfoLevel( InfoXMod, 0 );

7.2 Induced crossed modules

7.2.1 InducedXMod

. InducedXMod(args) (function)

. IsInducedXMod(xmod) (property)

. SurjectiveInducedXMod(xmod, hom) (operation)

. InclusionInducedXModByCopower(xmod, hom, list) (operation)

. MorphismOfInducedXMod(xmod) (attribute)

A morphism of crossed modules (σ ,ρ) : X1→X2 factors uniquely through an induced crossed
module ρ∗X1 = (δ : ρ∗S1 → R2). Similarly, a morphism of cat1-groups factors through an induced
cat1-group. Calculation of induced crossed modules of X also provides an algebraic means of deter-
mining the homotopy 2-type of homotopy pushouts of the classifying space of X . For more back-
ground from algebraic topology see references in [BH78], [BW95], [BW96]. Induced crossed modules
and induced cat1-groups also provide the building blocks for constructing pushouts in the categories
XMod and Cat1.

Data for the cases of algebraic interest is provided by a conjugation crossed module X = (∂ :
S→ R) and a homomorphism ι from R to a third group Q. (It is hoped to implement more general
cases in due course.) The output from the calculation is a crossed module ι∗X = (δ : ι∗S→ Q)
together with a morphism of crossed modules X → ι∗X . When ι is a surjection with kernel K then
ι∗S = [S,K] (see [BH78]). When ι is an inclusion the induced crossed module may be calculated
using a copower construction [BW95] or, in the case when R is normal in Q, as a coproduct of crossed
modules ([BW96], but not yet implemented). When ι is neither a surjection nor an inclusion, ι is
factored as the composite of the surjection onto the image and the inclusion of the image in Q, and then
the composite induced crossed module is constructed. These constructions use Tietze transformation
routines in the library file tietze.gi.

As a first, surjective example, we take for X the normal inclusion crossed module of a4 in s4,
and for ι the surjection from s4 to s3 with kernel k4. The induced crossed module is isomorphic to
X3.

Example

gap> s4gens := GeneratorsOfGroup( s4 );
[ (1,2), (2,3), (3,4) ]
gap> a4gens := GeneratorsOfGroup( a4 );
[ (1,2,3), (2,3,4) ]
gap> s3b := Group( (5,6),(6,7) );; SetName( s3b, "s3b" );
gap> epi := GroupHomomorphismByImages( s4, s3b, s4gens, [(5,6),(6,7),(5,6)] );;
gap> X4 := XModByNormalSubgroup( s4, a4 );;

gap> indX4 := SurjectiveInducedXMod( X4, epi );
[a4/ker->s3b]
gap> Display( indX4 );

Crossed module [a4/ker->s3b] :-
: Source group a4/ker has generators:

[ (1,3,2), (1,2,3) ]
: Range group s3b has generators:

[ (5,6), (6,7) ]
: Boundary homomorphism maps source generators to:

[ (5,6,7), (5,7,6) ]
: Action homomorphism maps range generators to automorphisms:

(5,6) --> { source gens --> [ (1,2,3), (1,3,2) ] }
(6,7) --> { source gens --> [ (1,2,3), (1,3,2) ] }
These 2 automorphisms generate the group of automorphisms.

gap> morX4 := MorphismOfInducedXMod( indX4 );
[[a4->s4] => [a4/ker->s3b]]

For a second, injective example we take for X a conjugation crossed module.
Example

gap> d8 := Subgroup( d16, [ b1^2, b2 ] ); SetName( d8, "d8" );
Group([ (11,13,15,17)(12,14,16,18), (12,18)(13,17)(14,16) ])
gap> c4 := Subgroup( d8, [ b1^2 ] ); SetName( c4, "c4" );
Group([ (11,13,15,17)(12,14,16,18) ])
gap> Y16 := XModByNormalSubgroup( d16, d8 );
[d8->d16]
gap> Y8 := SubXMod( Y16, c4, d8 );
[c4->d8]
gap> inc8 := InclusionMorphism2DimensionalDomains( Y16, Y8 );
[[c4->d8] => [d8->d16]]
gap> incd8 := RangeHom( inc8 );;
gap> indY8 := InducedXMod( Y8, incd8 );
i*([c4->d8])
gap> StructureDescription( indY8 );
[ "C4 x C4", "D16" ]
gap> morY8 := MorphismOfInducedXMod( indY8 );
[[c4->d8] => i*([c4->d8])]
gap> s3c := Subgroup( s4, [ (2,3), (3,4) ] );;
gap> SetName( s3c, "s3c" );
gap> indXs3c := InducedXMod( s4, s3c, s3c );
i*([s3c->s3c])
gap> StructureDescription( indXs3c );
[ "GL(2,3)", "S4" ]

For a third example we take the identity mapping on s3c as boundary, and the inclusion of s3c in s4
as ι . The induced group is a general linear group GL(2,3).

Example

gap> s3c := Subgroup( s4, [ (2,3), (3,4) ] );;
gap> SetName( s3c, "s3c" );
gap> indXs3c := InducedXMod( s4, s3c, s3c );
#I induced group has Size: 48
i*([s3c->s3c])
gap> StructureDescription( indXs3c );
[ "GL(2,3)", "S4" ]

7.2.2 AllInducedXMods

. AllInducedXMods(Q) (operation)

This function calculates all the induced crossed modules InducedXMod( Q, P, M ), where P
runs over all conjugacy classes of subgroups of Q and M runs over all non-trivial subgroups of P.

Example

gap> all := AllInducedXMods( q8 );;
gap> ids := List( all, x -> IdGroup(x) );;
gap> Sort( ids );
gap> ids;
[ [ [ 1, 1 ], [ 8, 4 ] ], [ [ 1, 1 ], [ 8, 4 ] ], [ [ 1, 1 ], [ 8, 4 ] ],

[ [ 1, 1 ], [ 8, 4 ] ], [ [ 4, 2 ], [ 8, 4 ] ], [ [ 4, 2 ], [ 8, 4 ] ],
[ [ 4, 2 ], [ 8, 4 ] ], [ [ 16, 2 ], [ 8, 4 ] ], [ [ 16, 2 ], [ 8, 4 ] ],
[ [ 16, 2 ], [ 8, 4 ] ], [ [ 16, 14 ], [ 8, 4 ] ] ]

7.3 Induced cat1-groups

7.3.1 InducedCat1Group

. InducedCat1Group(args) (function)

. InducedCat1GroupByFreeProduct(grp, hom) (property)

This area requires development.



Chapter 8

3d-groups and 3d-mappings : crossed
squares and cat2-groups

The term 3d-group refers to a set of equivalent categories of which the most common are the categories
of crossed squares and cat2-groups.

8.1 Definition of a crossed square and a crossed n-cube of groups

Crossed squares were introduced by Guin-Waléry and Loday (see, for example, [BL87]) as funda-
mental crossed squares of commutative squares of spaces, but are also of purely algebraic interest. We
denote by [n] the set {1,2, . . . ,n}. We use the n = 2 version of the definition of crossed n-cube as given
by Ellis and Steiner [ES87].

A crossed square S consists of the following:

• groups SJ for each of the four subsets J ⊆ [2];

• a commutative diagram of group homomorphisms:

∂̈1 : S[2]→ S{2}, ∂̈2 : S[2]→ S{1}, ∂̇1 : S{1}→ S /0, ∂̇2 : S{2}→ S /0;

• actions of S /0 on S{1},S{2} and S[2] which determine actions of S{1} on S{2} and S[2] via ∂̇1 and
actions of S{2} on S{1} and S[2] via ∂̇2 ;

• a function � : S{1}×S{2}→ S[2].

Here is a picture of the situation:

S[2]
∂̈1 //

∂̈2

��

S{2}

∂̇2

��

S =

S{1}
∂̇1

// S /0

The following axioms must be satisfied for all l ∈ S[2], m,m1,m2 ∈ S{1}, n,n1,n2 ∈ S{2}, p ∈ S /0:

51

• the homomorphisms ∂̈1, ∂̈2 preserve the action of S /0 ;

• each of the upper, left-hand, lower, and right-hand sides of the square,

S̈1 = (∂̈1 : S[2]→ S{2}),S̈2 = (∂̈2 : S[2]→ S{1}),Ṡ1 = (∂̇1 : S{1}→ S /0),Ṡ2 = (∂̇2 : S{2}→ S /0),

and the diagonal
S12 = (∂12 := ∂̇1∂̈2 = ∂̇2∂̈1 : S[2]→ S /0)

are crossed modules (with actions via S /0);

• � is a crossed pairing:

– (m1m2�n) = (m1�n)m2 (m2�n),

– (m�n1n2) = (m�n2) (m�n1)
n2 ,

– (m�n)p = (mp�np);

• ∂̈1(m�n) = (n−1)m n and ∂̈2(m�n) = m−1 mn,

• (m� ∂̈1l) = (l−1)m l and (∂̈2l�n) = l−1 ln.

Note that the actions of S{1} on S{2} and S{2} on S{1} via S /0 are compatible since

m1
(nm) = m1

∂̇2(nm) = m1
m−1(∂̇2n)m = ((m1

m−1
)n)m.

(A precrossed square is a similar structure which satisfies some subset of these axioms. [More
needed here.])

In what follows we shall generally use the following notation for the SJ , namely L = S[2]; M =
S{1}; N = S{2} and P = S /0.

Crossed squares are the n = 2 case of a crossed n-cube of groups, defined as follows. (This is an
attempt to translate Definition 2.1 in Ronnie’s Computing homotopy types using crossed n-cubes of
groups into right actions – but this definition is not yet completely understood!)

A crossed n-cube of groups consists of the following:

• groups SA for every subset A⊆ [n];

• a commutative diagram of group homomorphisms ∂i : SA → SA\{i}, i ∈ [n]; with composites
∂B : SA→ SA\B, B⊆ [n];

• actions of S /0 on each SA; and hence actions of SB on SA via ∂B for each B⊆ [n];

• functions �A,B : SA×SB→ SA∪B,(A,B⊆ [n]).

The following axioms must be satisfied (long list to be added).

8.2 Constructions for crossed squares

Analogously to the data structure used for crossed modules, crossed squares are implemented as
3d-groups. When times allows, cat2-groups will also be implemented, with conversion between the
two types of structure. Some standard constructions of crossed squares are listed below. At present,
a limited number of constructions are implemented. Morphisms of crossed squares have also been
implemented, though there is a lot still to be done.

8.2.1 CrossedSquare

. CrossedSquare(args) (function)

. CrossedSquareByNormalSubgroups(P, N, M, L) (operation)

. ActorCrossedSquare(X0) (operation)

. Transpose3dGroup(S0) (attribute)

. Name(S0) (attribute)

Here are some standard examples of crossed squares.

• If M,N are normal subgroups of a group P, and L=M∩N, then the four inclusions, L→N, L→
M, M→ P, N→ P, together with the actions of P on M,N and L given by conjugation, form a
crossed square with crossed pairing

� : M×N→M∩N, (m,n) 7→ [m,n] = m−1n−1mn = (n−1)mn = m−1mn .

This construction is implemented as CrossedSquareByNormalSubgroups(P,N,M,L);.

• The actor A (X0) of a crossed module X0 has been described in Chapter 5. The crossed pairing
is given by

� : R×W → S, (r,χ) 7→ χr .

This is implemented as ActorCrossedSquare( X0 );.

• The transpose of S is the crossed square S̃ obtained by interchanging S{1} with S{2}, ∂̈1 with
∂̈2, and ∂̇1 with ∂̇2. The crossed pairing is given by

�̃ : S{2}×S{1}→ S[2], (n,m) 7→ n�̃m := (m�n)−1 .

Example

gap> d20 := DihedralGroup( IsPermGroup, 20 );;
gap> gend20 := GeneratorsOfGroup( d20 );
[ (1,2,3,4,5,6,7,8,9,10), (2,10)(3,9)(4,8)(5,7) ]
gap> p1 := gend20[1];; p2 := gend20[2];; p12 := p1*p2;
(1,10)(2,9)(3,8)(4,7)(5,6)
gap> d10a := Subgroup( d20, [ p1^2, p2 ] );;
gap> d10b := Subgroup( d20, [ p1^2, p12 ] );;
gap> c5d := Subgroup( d20, [ p1^2 ] );;
gap> SetName( d20, "d20" ); SetName( d10a, "d10a" );
gap> SetName( d10b, "d10b" ); SetName( c5d, "c5d" );
gap> XSconj := CrossedSquareByNormalSubgroups( d20, d10b, d10a, c5d );
[ c5d -> d10b ]
[ | | ]
[ d10a -> d20 ]

gap> Name( XSconj );
"[c5d->d10b,d10a->d20]"
gap> XStrans := Transpose3dGroup( XSconj );
[ c5d -> d10a ]
[ | | ]
[ d10b -> d20 ]

gap> X20 := XModByNormalSubgroup( d20, d10a );
[d10a->d20]
gap> XSact := ActorCrossedSquare( X20 );
crossed square with:

up = Whitehead[d10a->d20]
left = [d10a->d20]
down = Norrie[d10a->d20]

right = Actor[d10a->d20]

8.2.2 CentralQuotient (for crossed modules)

. CentralQuotient(X0) (attribute)

The central quotient of a crossed module X = (∂ : S→ R) is the crossed square where:

• the left crossed module is X ;

• the right crossed module is the quotient X /Z(X ) (see CentreXMod (4.1.7));

• the top and bottom homomorphisms are the natural homomorphisms onto the quotient groups;

• the crossed pairing � : (R× F)→ S, where F = Fix(X ,S,R), is the displacement element
�(r,Fs) = 〈r,s〉= (s−1)rs (see Displacement (4.1.3) and section 4.3).

This is the special case of an intended function CrossedSquareByCentralExtension which has not
yet been implemented. In the example Xn7 E X24, constructed in section 4.1.

Example

gap> pos7 := Position( ids, [ [12,2], [24,5] ] );;
gap> Xn7 := nsx[pos7];
[Group( [ f2, f3, f4 ] )->Group( [ f1, f2, f4, f5 ] )]
gap> IdGroup( CentreXMod(Xn7) );
[ [ 4, 1 ], [ 4, 1 ] ]
gap> CQXn7 := CentralQuotient( Xn7 );
crossed square with:

up = [Group( [ f2, f3, f4 ] )->Group( [ f1 ] )]
left = [Group( [ f2, f3, f4 ] )->Group( [ f1, f2, f4, f5 ] )]
down = [Group( [ f1, f2, f4, f5 ] )->Group( [ f1, f2 ] )]

right = [Group( [ f1 ] )->Group( [ f1, f2 ] )]

gap> IdGroup( CQXn7 );
[ [ [ 12, 2 ], [ 3, 1 ] ], [ [ 24, 5 ], [ 6, 1 ] ] ]

8.2.3 IsCrossedSquare

. IsCrossedSquare(obj) (property)

. Is3dObject(obj) (property)

. IsPerm3dObject(obj) (property)

. IsPc3dObject(obj) (property)



XMod 55

. IsFp3dObject(obj) (property)

. IsPreCrossedSquare(obj) (property)

These are the basic properties for 3d-groups, and crossed squares in particular.

8.2.4 Up2DimensionalGroup

. Up2DimensionalGroup(XS) (attribute)

. Left2DimensionalGroup(XS) (attribute)

. Down2DimensionalGroup(XS) (attribute)

. Right2DimensionalGroup(XS) (attribute)

. DiagonalAction(XS) (attribute)

. CrossedPairing(XS) (attribute)

. ImageElmCrossedPairing(XS, pair) (operation)

In this implementation the attributes used in the construction of a crossed square XS are the four
crossed modules (2d-groups) on the sides of the square (up; down, left; and right); the diagonal action
of P on L; and the crossed pairing.

The GAP development team have suggested that crossed pairings should be implemented as a
special case of BinaryMappings – a structure which does not yet exist in GAP. As a temporary
measure, crossed pairings have been implemented using Mapping2ArgumentsByFunction.

Example

gap> Up2DimensionalGroup( XSconj );
[c5d->d10b]
gap> Right2DimensionalGroup( XSact );
Actor[d10a->d20]
gap> xpconj := CrossedPairing( XSconj );;
gap> ImageElmCrossedPairing( xpconj, [ p2, p12 ] );
(1,9,7,5,3)(2,10,8,6,4)
gap> diag := DiagonalAction( XSact );
[ (1,3,5,2,4)(6,10,14,8,12)(7,11,15,9,13), (1,2,5,4)(6,8,14,12)(7,11,13,9)
] ->

[ (1,3,5,2,4)(6,10,14,8,12)(7,11,15,9,13), (1,2,5,4)(6,8,14,12)(7,11,13,9)
] -> [ ^(1,3,5,7,9)(2,4,6,8,10), ^(1,2,5,4)(3,8)(6,7,10,9) ]

8.3 Morphisms of crossed squares

This section describes an initial implementation of morphisms of (pre-)crossed squares.

8.3.1 Source

. Source(map) (attribute)

. Range(map) (attribute)

. Up2DimensionalMorphism(map) (attribute)

. Left2DimensionalMorphism(map) (attribute)

. Down2DimensionalMorphism(map) (attribute)

. Right2DimensionalMorphism(map) (attribute)

Morphisms of 3dObjects are implemented as 3dMappings. These have a pair of 3d-groups
as source and range, together with four 2d-morphisms mapping between the four pairs of crossed
modules on the four sides of the squares. These functions return fail when invalid data is supplied.

8.3.2 IsCrossedSquareMorphism

. IsCrossedSquareMorphism(map) (property)

. IsPreCrossedSquareMorphism(map) (property)

. IsBijective(mor) (method)

. IsEndomorphism3dObject(mor) (property)

. IsAutomorphism3dObject(mor) (property)

A morphism mor between two pre-crossed squares S1 and S2 consists of four crossed
module morphisms Up2DimensionalMorphism(mor), mapping the Up2DimensionalGroup of
S1 to that of S2, Left2DimensionalMorphism(mor), Down2DimensionalMorphism(mor) and
Right2DimensionalMorphism(mor). These four morphisms are required to commute with
the four boundary maps and to preserve the rest of the structure. The current version of
IsCrossedSquareMorphism does not perform all the required checks.

Example

gap> ad20 := GroupHomomorphismByImages( d20, d20, [p1,p2], [p1,p2^p1] );;
gap> ad10a := GroupHomomorphismByImages( d10a, d10a, [p1^2,p2], [p1^2,p2^p1] );;
gap> ad10b := GroupHomomorphismByImages( d10b, d10b, [p1^2,p12], [p1^2,p12^p1] );;
gap> idc5d := IdentityMapping( c5d );;
gap> upconj := Up2DimensionalGroup( XSconj );;
gap> leftconj := Left2DimensionalGroup( XSconj );;
gap> downconj := Down2DimensionalGroup( XSconj );;
gap> rightconj := Right2DimensionalGroup( XSconj );;
gap> up := XModMorphismByHoms( upconj, upconj, idc5d, ad10b );
[[c5d->d10b] => [c5d->d10b]]
gap> left := XModMorphismByHoms( leftconj, leftconj, idc5d, ad10a );
[[c5d->d10a] => [c5d->d10a]]
gap> down := XModMorphismByHoms( downconj, downconj, ad10a, ad20 );
[[d10a->d20] => [d10a->d20]]
gap> right := XModMorphismByHoms( rightconj, rightconj, ad10b, ad20 );
[[d10b->d20] => [d10b->d20]]
gap> autoconj := CrossedSquareMorphism( XSconj, XSconj, up, left, right, down );;
gap> ord := Order( autoconj );;
gap> Display( autoconj );
Morphism of crossed squares :-
: Source = [c5d->d10b,d10a->d20]
: Range = [c5d->d10b,d10a->d20]
: order = 5
: up-left: [ [ ( 1, 3, 5, 7, 9)( 2, 4, 6, 8,10) ],

[ ( 1, 3, 5, 7, 9)( 2, 4, 6, 8,10) ] ]
: up-right:
[ [ ( 1, 3, 5, 7, 9)( 2, 4, 6, 8,10), ( 1,10)( 2, 9)( 3, 8)( 4, 7)( 5, 6) ],

[ ( 1, 3, 5, 7, 9)( 2, 4, 6, 8,10), ( 1, 2)( 3,10)( 4, 9)( 5, 8)( 6, 7) ] ]



XMod 57

: down-left:
[ [ ( 1, 3, 5, 7, 9)( 2, 4, 6, 8,10), ( 2,10)( 3, 9)( 4, 8)( 5, 7) ],

[ ( 1, 3, 5, 7, 9)( 2, 4, 6, 8,10), ( 1, 3)( 4,10)( 5, 9)( 6, 8) ] ]
: down-right:
[ [ ( 1, 2, 3, 4, 5, 6, 7, 8, 9,10), ( 2,10)( 3, 9)( 4, 8)( 5, 7) ],

[ ( 1, 2, 3, 4, 5, 6, 7, 8, 9,10), ( 1, 3)( 4,10)( 5, 9)( 6, 8) ] ]
gap> IsAutomorphismHigherDimensionalDomain( autoconj );
true
gap> KnownPropertiesOfObject( autoconj );
[ "CanEasilyCompareElements", "CanEasilySortElements", "IsTotal",

"IsSingleValued", "IsInjective", "IsSurjective",
"IsPreCrossedSquareMorphism", "IsCrossedSquareMorphism",
"IsEndomorphismHigherDimensionalDomain",
"IsAutomorphismHigherDimensionalDomain" ]

8.4 Definitions and constructions for cat2-groups and their morphisms

We shall give three definitions of cat2-groups and show that they are equivalent. When we come to
define catn-groups we shall give a similar set of three definitions.

Firstly, we take the definition of a cat2-group from Section 5 of Brown and Loday [BL87], suitably
modified. A cat2-group C = (C[2],C{2},C{1},C/0) comprises four groups (one for each of the subsets
of [2]) and 15 homomorphisms, as shown in the following diagram:

C[2]

��

ẗ2,ḧ2

��

//
ẗ1,ḧ1 //

��

t[2],h[2]

��

C{2}ë1

oo

��

ṫ2,ḣ2

��

C =

C{1}

ë2

OO

//

ṫ1,ḣ1

//
C/0

ė2

OO

ė1oo

e[2]

__

The following axioms are satisfied by these homomorphisms:

• the four sides of the square are cat1-groups, denoted C̈1, C̈2, Ċ1, Ċ2,

• ṫ1◦ ḧ2 = ḣ2◦ ẗ1, ṫ2◦ ḧ1 = ḣ1◦ ẗ2, ė1◦ ṫ2 = ẗ2◦ ë1, ė2◦ ṫ1 = ẗ1◦ ë2, ė1◦ ḣ2 = ḧ2◦ ë1, ė2◦ ḣ1 = ḧ1◦ ë2,

• ṫ1 ◦ ẗ2 = ṫ2 ◦ ẗ1 = t[2], ḣ1 ◦ ḧ2 = ḣ2 ◦ ḧ1 = h[2], ė1 ◦ ë2 = ė2 ◦ ë1 = e[2], making the diagonal a
cat1-group (e[2]; t[2],h[2] : C[2]→C/0).

It follows from these identities that (ẗ1, ṫ1), (ḧ1, ḣ1) and (ë1, ė1) are morphisms of cat1-groups.
Secondly, we give the simplest of the three definitions, adapted from Ellis-Steiner [ES87]. A cat2-

group C consists of groups G,R1,R2 and six homomorphisms t1,h1 : G→ R2, e1 : R2 → G, t2,h2 :
G→ R1, e2 : R1→ G, satisfying the following axioms for all 16 i6 2,

• (ti ◦ ei)r = r, (hi ◦ ei)r = r, ∀r ∈ R[2]\{i}, [ker ti,kerhi] = 1,

• (e1 ◦ t1)◦ (e2 ◦ t2) = (e2 ◦ t2)◦ (e1 ◦ t1), (e1 ◦h1)◦ (e2 ◦h2) = (e2 ◦h2)◦ (e1 ◦h1),

• (e1 ◦ t1)◦ (e2 ◦h2) = (e2 ◦h2)◦ (e1 ◦ t1), (e2 ◦ t2)◦ (e1 ◦h1) = (e1 ◦h1)◦ (e2 ◦ t2).

Our third definition defines a cat2-group as a "cat1-group of cat1-groups". A cat2-group C consists
of two cat1-groups C1 = (e1; t1,h1 : G1 → R1) and C2 = (e2; t2,h2 : G2 → R2) and cat1-morphisms
t = (ẗ, ṫ), h = (ḧ, ḣ) : C1→ C2, e = (ë, ė) : C2→ C1, subject to the following conditions:

(t ◦ e) and (h◦ e) are the identity mapping on C2, [ker t,kerh] = {1C1},

where ker t = (ker ẗ, ker ṫ), and similarly for kerh.

8.4.1 Cat2Group

. Cat2Group(args) (function)

. PreCat2Group(args) (function)

. PreCat2GroupByPreCat1Groups(L) (operation)

The global functions Cat2Group and PreCat2Group are normally called with a single argument,
a list of cat1-groups.

Example

gap> CC6 := Cat2Group( Cat1Group(6,2,2), Cat1Group(6,2,3) );
generating (pre-)cat1-groups:
1 : [C6=>Group( [ f1 ] )]
2 : [C6=>Group( [ f2 ] )]

gap> IsCat2Group( CC6 );
true

8.4.2 Cat2GroupOfCrossedSquare

. Cat2GroupOfCrossedSquare(xsq) (attribute)

. CrossedSquareOfCat2Group(CC) (attribute)

These functions are very experimental, and should not be relied on!
These functions provide the conversion from crossed square to cat2-group, and conversely. (They

are the 3-dimensional equivalents of Cat1GroupOfXMod (2.5.2) and XModOfCat1Group (2.5.2).)
Example

gap> xsCC6 := CrossedSquareOfCat2Group( CC6 );
crossed square with:

up = [Group( () )->Group( [ (1,2) ] )]
left = [Group( () )->Group( [ (), (3,4,5) ] )]
down = [Group( [ (), (3,4,5) ] ) -> Group( () )]

right = [Group( [ (1,2) ] ) -> Group( () )]
gap> Cat2GroupOfCrossedSquare( XSact );

Warning: these conversion functions are still under development
fail

8.5 Definition and constructions for catn-groups and their morphisms

In this chapter we are interested in cat2-groups, but it is convenient in this section to give the more
general definition. There are three equivalent description of a catn-group.

A catn-group consists of the following.

• 2n groups GA, one for each subset A of [n], the vertices of an n-cube.

• Group homomorphisms forming n2n−1 commuting cat1-groups,

CA,i = (eA,i; tA,i, hA,i : GA→ GA\{i}), for all A⊆ [n], i ∈ A,

the edges of the cube.

• These cat1-groups combine (in sets of 4) to form n(n−1)2n−3 cat2-groups CA,{i, j} for all {i, j}⊆
A⊆ [n], i 6= j, the faces of the cube.

Note that, since the tA,i,hA,i and eA,i commute, composite homomorphisms tA,B,hA,B : GA→ GA\B and
eA,B : GA\B→ GA are well defined for all B⊆ A⊆ [n].

Secondly, we give the simplest of the three descriptions, again adapted from Ellis-Steiner [ES87].
A catn-group C consists of 2n groups GA, one for each subset A of [n], and 3n homomorphisms

t[n],i,h[n],i : G[n]→ G[n]\{i}, e[n],i : G[n]\{i}→ G[n],

satisfying the following axioms for all 16 i6 n,}

• the C[n],i = (e[n],i; t[n],i, h[n],i : G[n]→ G[n]\{i}) are commuting cat1-groups, so that:

• (e1 ◦ t1)◦ (e2 ◦ t2) = (e2 ◦ t2)◦ (e1 ◦ t1), (e1 ◦h1)◦ (e2 ◦h2) = (e2 ◦h2)◦ (e1 ◦h1),

• (e1 ◦ t1)◦ (e2 ◦h2) = (e2 ◦h2)◦ (e1 ◦ t1), (e2 ◦ t2)◦ (e1 ◦h1) = (e1 ◦h1)◦ (e2 ◦ t2).

Our third description defines a catn-group as a "cat1-group of cat(n−1)-groups".
A catn-group C consists of two cat(n−1)-groups:

• A with groups GA, A⊆ [n−1], and homomorphisms ẗA,i, ḧA,i, ëA,i,

• B with groups HB, B⊆ [n−1], and homomorphisms ṫB,i, ḣB,i, ėB,i, and

• cat(n−1)-morphisms t,h : A →B and e : B→A subject to the following conditions:

(t ◦ e) and (h◦ e) are the identity mapping on B, [ker t,kerh] = {1A }.

Crossed modules of groupoids

The material documented in this chapter is experimental, and is likely to be changed very soon.

9.1 Constructions for crossed modules of groupoids

A typical example of a crossed module X over a groupoid has for its range a connected groupoid.
This is a direct product of a group with a complete graph, and we call the vertices of the graph the
objects of the crossed module. The source of X is a groupoid, with the same objects, which is either
discrete or connected. The boundary morphism is constant on objects. For details and other references
see [AW10].

9.1.1 SinglePiecePreXModWithObjects

. SinglePiecePreXModWithObjects(pxmod, obs, isdisc) (operation)

At present the experimental operation SinglePiecePreXModWithObjects accepts a precrossed
module pxmod, a set of objects obs, and a boolean isdisc which is true when the source groupoid is
homogeneous and discrete and false when the source groupoid is connected. Other operations will
be added as time permits.

In the example the crossed module DX4 has discrete source, and is a groupoid equivalent of
XModByNormalSubgroup (2.1.1).

Example

gap> s4 := Group( (1,2,3,4), (3,4) );;
gap> SetName( s4, "s4" );
gap> a4 := Subgroup( s4, [ (1,2,3), (2,3,4) ] );;
gap> SetName( a4, "a4" );
gap> X4 := XModByNormalSubgroup( s4, a4 );;
gap> DX4 := SinglePiecePreXModWithObjects( X4, [-9,-8,-7], false );
precrossed module with source groupoid:
single piece groupoid: < a4, [ -9, -8, -7 ] >
and range groupoid:
single piece groupoid: < s4, [ -9, -8, -7 ] >
gap> Ga4 := Source( DX4 );;
gap> Gs4 := Range( DX4 );;

9.1.2 IsXModWithObjects

. IsXModWithObjects(pxmod) (property)

. IsPreXModWithObjects(pxmod) (property)

. IsDirectProductWithCompleteDigraphDomain(pxmod) (property)

The precrossed module DX4 belongs to the category Is2DimensionalGroupWithObjects and is,
of course, a crossed module.

Example

gap> IsXModWithObjects( DX4 );
true
gap> KnownPropertiesOfObject( DX4 );
[ "CanEasilyCompareElements", "CanEasilySortElements", "IsDuplicateFree",

"IsGeneratorsOfSemigroup", "IsSinglePieceDomain",
"IsDirectProductWithCompleteDigraphDomain", "IsPreXModWithObjects",
"IsXModWithObjects" ]

9.1.3 IsPermPreXModWithObjects

. IsPermPreXModWithObjects(pxmod) (property)

. IsPcPreXModWithObjects(pxmod) (property)

. IsFpPreXModWithObjects(pxmod) (property)

To test these properties we test the precrossed modules from which they were constructed.
Example

gap> IsPermPreXModWithObjects( DX4 );
true
gap> IsPcPreXModWithObjects( DX4 );
false
gap> IsFpPreXModWithObjects( DX4 );
false

9.1.4 Root2dGroup

. Root2dGroup(pxmod) (attribute)

. XModAction(pxmod) (attribute)

The attributes of a precrossed module with objects include the standard Source; Range; Boundary
(2.1.2); and XModAction (2.1.2) as with precrossed modules of groups. There is also ObjectList,
as in the groupoids package. Additionally there is Root2dGroup which is the underlying precrossed
module used in the construction.

Note that XModAction is now a groupoid homomorphism from the source groupoid to a one-object
groupoid (with object 0) where the group is the automorphism group of the range groupoid.

Example

gap> KnownAttributesOfObject(DX4);
[ "Range", "Source", "Boundary", "ObjectList", "XModAction", "Root2dGroup" ]
gap> Root2dGroup( DX4 );
[a4->s4]
gap> act := XModAction( DX4 );;
gap> r := Arrow( Gs4, (1,2,3,4), -7, -8 );;
gap> ImageElm( act, r );
[groupoid homomorphism :
[ [ [(1,2,3) : -9 -> -9], [(2,3,4) : -9 -> -9], [() : -9 -> -8],

[() : -9 -> -7] ],
[ [(2,3,4) : -9 -> -9], [(1,3,4) : -9 -> -9], [() : -9 -> -7],

[() : -9 -> -8] ] ] : 0 -> 0]
gap> s := Arrow( Ga4, (1,2,4), -8, -8 );;
gap> ## calculate s^r
gap> ims := ImageElmXModAction( DX4, s, r );
[(1,2,3) : -7 -> -7]

There is much more to be done with these constructions.

Applications

This chapter was added in April 2018 for version 2.66 of XMod. Initially it describes crossed modules
for free loop spaces. Further applications may arise in due course.

10.1 Free Loop Spaces

These functions have been used to produce examples for Ronald Brown’s paper Crossed modules, and
the homotopy 2-type of a free loop space [Bro18]. The relevant theorem in that paper is as follows.

THEOREM 2.1 Let M = (∂ : M → P) be a crossed module of groups and let X = BM be the
classifying space of M . Then the components of LX, the free loop space on X, are determined by
equivalence classes of elements a ∈ P where a,a′ are equivalent if and only if there are elements
m ∈M, p ∈ P such that a′ = p+a−∂m− p.

Further the homotopy 2-type of a component of LX given by a ∈ P is determined by the crossed
module of groups LM [a] = (∂a : M→ P(a)) where:

• P(a) is the subgroup of the cat1-group G = PnM such that ∂m = [p,a] =−p−a+ p+a;

• ∂a(m) = (∂m,m−1ma) for m ∈M;

• the action of P(a) on M is given by n(p,m) = np for n ∈M, (p,m) ∈ P(a).

In particular π1(LX ,a) is isomorphic to cokernel(∂a), and π2(LX ,a)∼= π2(X ,∗)ā, the elements of
π2(X ,∗) fixed under the action of ā, the class of a in π1(X ,∗).

There is an exact sequence π
φ→ π→ π1(LX ,a)→Cā(π1(X ,∗))→ 1, in which π = π2(X ,∗), and

φ is the morphism m 7→ m−1ma.

10.1.1 LoopsXMod

. LoopsXMod(M, a) (operation)

. AllLoopsXMod(M) (operation)

The operation LoopsXMod(M,a) calculates the crossed module LM [a] described in the theorem.
The operation AllLoopsXMod(M) returns a list of crossed modules, one for each equivalence class

of elements p ∈ P. THESE OPERATIONS SHOULD BE CONSIDERED EXPERIMENTAL AT PRESENT.

In the example below the automorphism crossed module X8 has M ∼=C3
2 and P = PSL(3,2) is the

automorphism group of M. There are 6 equivalence classes and, for each LX calculated, the Size
(2.1.4) and StructureDescription (2.7.1) are printed out.

Example

gap> k8 := Group( (3,4), (5,6), (7,8) );;
gap> SetName( k8, "k8" );
gap> Y8 := XModByAutomorphismGroup( k8 );;
gap> X8 := Image( IsomorphismPerm2DimensionalGroup( Y8 ) );;
gap> SetName( X8, "X8" );
gap> Print( "X8: ", Size( X8 ), " : ", StructureDescription( X8 ), "\n" );
X8: [ 8, 168 ] : [ "C2 x C2 x C2", "PSL(3,2)" ]
gap> LX := LoopsXMod( X8, (1,2)(5,6) );;
gap> Size( LX ); StructureDescription( LX );
[ 8, 64 ]
[ "C2 x C2 x C2", "((C2 x C2 x C2 x C2) : C2) : C2" ]
gap> SetInfoLevel( InfoXMod, 1 );
gap> LX8 := AllLoopsXMod( X8 );;
#I LoopsXMod with a = (), [ 8, 1344 ]
#I LoopsXMod with a = (4,5)(6,7), [ 8, 64 ]
#I LoopsXMod with a = (2,3)(4,6,5,7), [ 8, 32 ]
#I LoopsXMod with a = (2,4,6)(3,5,7), [ 8, 24 ]
#I LoopsXMod with a = (1,2,4,3,6,7,5), [ 8, 56 ]
#I LoopsXMod with a = (1,2,4,5,7,3,6), [ 8, 56 ]
gap> iso := IsomorphismGroups( Range( LX ), Range( LX8[2] ) );
[ (1,2)(3,4)(5,6)(7,8), (1,3)(2,4)(5,7)(6,8), (1,5)(2,6)(3,7)(4,8),

(5,8)(6,7), (2,3)(6,7), (2,7)(3,6) ] ->
[ (1,5)(2,6)(3,7)(4,8), (1,6)(2,5)(3,8)(4,7), (1,4)(2,3)(5,8)(6,7),

(1,2)(5,6), (1,2)(3,4), (1,3)(2,4) ]

Utility functions

By a utility function we mean a GAP function which is

• needed by other functions in this package,

• not (as far as we know) provided by the standard GAP library,

• more suitable for inclusion in the main library than in this package.

Sections on Printing Lists and Distinct and Common Representatives were moved to the Utils package
with version 2.56.

11.1 Inclusion and Restriction Mappings

These functions have been moved to the gpd package, but are still documented here.

11.1.1 InclusionMappingGroups

. InclusionMappingGroups(G, H) (operation)

. MappingToOne(G, H) (operation)

This set of utilities concerns mappings. The map incd8 is the inclusion of d8 in d16 used in
Section 3.4. MappingToOne(G,H) maps the whole of G to the identity element in H.

Example

gap> Print( incd8, "\n" );
[ (11,13,15,17)(12,14,16,18), (11,18)(12,17)(13,16)(14,15) ] ->
[ (11,13,15,17)(12,14,16,18), (11,18)(12,17)(13,16)(14,15) ]
gap> imd8 := Image( incd8 );;
gap> MappingToOne( c4, imd8 );
[ (11,13,15,17)(12,14,16,18) ] -> [ () ]

11.1.2 InnerAutomorphismsByNormalSubgroup

. InnerAutomorphismsByNormalSubgroup(G, N) (operation)

. IsGroupOfAutomorphisms(A) (property)

Inner automorphisms of a group G by the elements of a normal subgroup N are calculated with the
first of these functions, usually with G = N.

Example

gap> autd8 := AutomorphismGroup( d8 );;
gap> innd8 := InnerAutomorphismsByNormalSubgroup( d8, d8 );;
gap> GeneratorsOfGroup( innd8 );
[ ^(1,2,3,4), ^(1,3) ]
gap> IsGroupOfAutomorphisms( innd8 );
true

11.2 Abelian Modules

11.2.1 AbelianModuleObject

. AbelianModuleObject(grp, act) (operation)

. IsAbelianModule(obj) (property)

. AbelianModuleGroup(obj) (attribute)

. AbelianModuleAction(obj) (attribute)

An abelian module is an abelian group together with a group action. These are used by the crossed
module constructor XModByAbelianModule (2.1.1).

The resulting Xabmod is isomorphic to the output from XModByAutomorphismGroup( k4 );.
Example

gap> x := (6,7)(8,9);; y := (6,8)(7,9);; z := (6,9)(7,8);;
gap> k4a := Group( x, y );; SetName( k4a, "k4a" );
gap> gens3a := [ (1,2), (2,3) ];;
gap> s3a := Group( gens3a );; SetName( s3a, "s3a" );
gap> alpha := GroupHomomorphismByImages( k4a, k4a, [x,y], [y,x] );;
gap> beta := GroupHomomorphismByImages( k4a, k4a, [x,y], [x,z] );;
gap> auta := Group( alpha, beta );;
gap> acta := GroupHomomorphismByImages( s3a, auta, gens3a, [alpha,beta] );;
gap> abmod := AbelianModuleObject( k4a, acta );;
gap> Xabmod := XModByAbelianModule( abmod );
[k4a->s3a]
gap> Display( Xabmod );

Crossed module [k4a->s3a] :-
: Source group k4a has generators:

[ (6,7)(8,9), (6,8)(7,9) ]
: Range group s3a has generators:

[ (1,2), (2,3) ]
: Boundary homomorphism maps source generators to:

[ (), () ]
: Action homomorphism maps range generators to automorphisms:

(1,2) --> { source gens --> [ (6,8)(7,9), (6,7)(8,9) ] }
(2,3) --> { source gens --> [ (6,7)(8,9), (6,9)(7,8) ] }
These 2 automorphisms generate the group of automorphisms.

Development history

This chapter, which contains details of the major changes to the package as it develops, was first
created in April 2002. Details of the changes from XMod 1 to XMod 2.001 are far from complete.
Starting with version 2.009 the file CHANGES lists the minor changes as well as the more fundamental
ones.

The inspiration for this package was the need, in the mid-1990’s, to calculate induced crossed
modules (see [BW95], [BW96], [BW03]). GAP was chosen over other computational group theory
systems because the code was freely available, and it was possible to modify the Tietze transformation
code so as to record the images of the original generators of a presentation as words in the simplified
presentation. (These modifications are now a standard part of the Tietze transformation package in
GAP.)

12.1 Changes from version to version

12.1.1 Version 1 for GAP 3

The first version of XMod became an accepted package for GAP 3.4.3 in December 1996.

12.1.2 Version 2

Conversion of XMod 1 from GAP 3.4.3 to the new GAP syntax began soon after GAP 4 was released,
and had a lengthy gestation. The new GAP syntax encouraged a re-naming of many of the function
names. An early decision was to introduce generic categories 2dDomain for (pre-)crossed modules
and (pre-)cat1-groups, and 2dMapping for the various types of morphism. In 2.009 3dDomain was
used for crossed squares and cat2-groups, and 3dMapping for their morphisms. A generic name for
derivations and sections is also required, and Up2dMapping is currently used.

12.1.3 Version 2.001 for GAP 4

This was the first version of XMod for GAP 4, completed in April 2002 in time for the release of GAP
4.3. Functions for actors and induced crossed modules were not included, nor many of the functions
for derivations and sections, for example InnerDerivation.

12.1.4 Induced crossed modules

During May 2002 converted the code for induced crossed modules. (Induced cat1-groups may be
converted one day.)

12.1.5 Versions 2.002 – 2.006

Version 2.004 of April 14th 2004 added the Cat1Select (2.6.1) functionality of version 1 to the
Cat1Group (2.4.1) function.

A significant addition in Version 2.005 was the conversion of the actor crossed module func-
tions from the 3.4.4 version. This included AutomorphismPermGroup (6.1.1) for a crossed module;
WhiteheadXMod (6.1.2); NorrieXMod (6.1.2); LueXMod (6.1.2); ActorXMod (6.1.2); CentreXMod
(4.1.7) of a crossed module; InnerMorphism (6.1.3); and InnerActorXMod (6.1.3).

12.1.6 Versions 2.007 – 2.010

These versions contain changes made between September 2004 and October 2007.

• Added basic functions for crossed squares, considered as 3dObjects with crossed pairings,
and their morphisms. Groups with two normal subgroups, and the actor of a crossed module,
provide standard examples of crossed squares. (Cat2-groups are not yet implemented.)

• Converted the documentation to the format of the GAPDoc package.

• Improved AutomorphismPermGroup (6.1.1) for crossed modules, and introduced a special
method for conjugation crossed modules.

• Substantial revisons made to XModByCentralExtension (2.1.1); NorrieXMod (6.1.2);
LueXMod (6.1.2); ActorXMod (6.1.2); and InclusionInducedXModByCopower (7.2.1).

• Version 2.010, of October 2007, was timed to coincide with the release of GAP 4.4.10, and
included a change of filenames; and correct file protection codes.

12.2 Versions for GAP [4.5 .. 4.8]

Version 2.19, released on 9th June 2012, included the following changes:

• The file makedocrel.g was copied, with appropriate changes, from GAPDoc, and now pro-
vides the correct way to update the documentation.

• The first functions for crossed modules of groupoids were introduced.

• A GNU General Public License declaration was added.

12.2.1 AllCat1s

Version 2.21 contained major changes to the Cat1Select (2.6.1) operation: the list CAT1_LIST of
cat1-structures in the data file cat1data.g was changed from permutation groups to pc-groups, with
the generators of subgroups; images of the tail map; and images of the head map being given as
ExtRepOfObj of words in the generators.



XMod 70

The AllCat1s function was reintroduced from the GAP3 version, and renamed as the operation
AllCat1DataGroupsBasic (2.6.2).

In version 2.25 the data in cat1data.g was extended from groups of size up to 48 to groups of
size up to 70. In particular, the 267 groups of size 64 give rise to a total of 1275 cat1-groups. The
authors are indebted to Van Luyen Le in Galway for pointing out a number of errors in the version of
this list distributed with version 2.24 of this package.

12.2.2 Versions 2.43 - 2.56

Version 2.43, released on 11th November 2015, included the following changes:

• The material on isoclinism in Chapter 4 was added.

• The package webpage has moved to http://pages.bangor.ac.uk/~mas023/chda/.

• A GitHub repository was started at: https://github.com/gap-packages/xmod.

• The section on Distinct and Common Representatives was moved to the Utils package.

12.2.3 Version 2.61

Major changes in names took place, with 2dDomain, 2dGroup, 2dMapping, etc. becoming
2DimensionalDomain, 2DimensionalGroup, 2DimensionalMapping, etc., and similarly for 3-
dimensional versions. Also HigherDimensionalDomain and related categories, domains, properties,
attributes and operations were introduced. At the same time, functions for cat2-groups were intro-
duced by Alper Odabas.

12.2.4 Latest Version

The latest version conmtains additional material on crossed modules of groupoids.

12.3 What needs doing next?

• Speed up the calculation of Whitehead groups.

• Add more functions for 3dObjects and implement cat2-groups.

• Improve interaction with the package groupoids implementing the group groupoid version of a
crossed module, and adding more functions for crossed modules of groupoids.

• Add interaction with IdRel (and possibly XRes and natp) .

• Need InverseGeneralMapping for morphisms and more features for FpXMods, PcXMods, etc.

• Implement actions of a crossed module.

• Implement FreeXMods and an operation Isomorphism2dDomains.

• Allow the construction of a group of morphisms of crossed modules.

• Complete the conversion from Version 1 of the calculation of sections using EndoClasses.

http://pages.bangor.ac.uk/~mas023/chda/
https://github.com/gap-packages/xmod


XMod 71

• More crossed square constructions:

– If M,N are ordinary P-modules and A is an arbitrary abelian group on which P acts triv-
ially, then there is a crossed square with sides

0 : A→ N, 0 : A→M, 0 : M→ P, 0 : N→ P.

– For a group L, the automorphism crossed module Act L = (ι : L→ Aut L) splits to form
the square with (ι1 : L→ Inn L) on two sides, and (ι2 : Inn L→ Aut L) on the other two
sides, where ι1 maps l ∈ L to the inner automorphism βl : L→ L, l′ 7→ l−1l′l, and ι2 is the
inclusion of Inn L in Aut L. The actions are standard, and the crossed pairing is

� : Inn L× Inn L→ L, (βl,βl′) 7→ [l, l′] .

• Improve the interaction with the HAP package.

for precat1-morphisms, 24

IdentitySection, 39
IdGroup

for 2d-groups, 20
for crossed modules, 9

ImageElmCrossedPairing, 55
ImageElmXModAction, 9
ImagesList, 41
ImagesTable, 40
inclusion mapping, 65
InclusionInducedXModByCopower, 48
InclusionMappingGroups, 65
InclusionMorphism2DimensionalDomains

for cat1-groups, 24
for crossed modules, 22

induced cat1-groups, 50
induced crossed module, 48
InducedCat1Group, 50
InducedCat1GroupByFreeProduct, 50
InducedXMod, 48
InfoXMod, 6
InnerActorXMod, 46
InnerAutomorphismCat1, 24
InnerAutomorphismsByNormalSubgroup, 66
InnerAutomorphismXMod, 22
InnerMorphism, 46
IntersectionSubXMods, 28
Is2DimensionalDomain, 10
Is2DimensionalGroup, 10
Is2DimensionalGroupWithObjects, 61
Is3dObject, 54

IsAbelian2DimensionalGroup, 31
IsAbelianModule, 66
IsAbelianModule2DimensionalGroup, 10
IsAspherical2DimensionalGroup, 31
IsAutomorphism3dObject, 56
IsAutomorphismGroup2DimensionalGroup, 10
IsBijective, 56

for pre-xmod morphisms, 22
IsCat1Group, 16
IsCat1Morphism, 24
IsCentralExtension2DimensionalGroup, 10
IsCrossedSquare, 54
IsCrossedSquareMorphism, 56
IsDerivation, 37
IsDirectProductWithCompleteDigraph-

Domain, 61
IsEndo2DimensionalMapping, 22
IsEndomorphism3dObject, 56
IsEndomorphismPreCat1Group, 16
IsFaithful2DimensionalGroup, 31
IsFp2DimensionalGroup, 11
IsFp3dObject, 55
IsFpPreXModWithObjects, 61
IsGroupOfAutomorphisms, 66
IsIdentityCat1Group, 16
IsInducedXMod, 48
IsInjective

for pre-xmod morphisms, 22
IsMonoidOfUp2DimensionalMappingsObj, 40
IsNilpotent2DimensionalGroup, 31
IsNormal for crossed modules, 11
IsNormalSubgroup2DimensionalGroup, 10
IsoclinicMiddleLength

for crossed modules of groups, 36
for groups, 34

IsoclinicRank
for crossed modules of groups, 36
for groups, 34

IsoclinicStemDomain
for crossed modules of groups, 35
for groups, 34

Isoclinism
for crossed modules, 35
for groups, 33

IsomorphismByIsomorphisms, 23
IsomorphismFp2DimensionalGroup

for pre-cat1 morphisms, 24
IsomorphismPc2DimensionalGroup

for pre-cat1 morphisms, 24
for pre-xmod morphisms, 23

IsomorphismPerm2DimensionalGroup
for pre-cat1 morphisms, 24
for pre-xmod morphisms, 23

IsomorphismPermObject, 24
IsomorphismXMods, 33
IsPc2DimensionalGroup, 11
IsPc3dObject, 54
IsPcPreXModWithObjects, 61
IsPerm2DimensionalGroup, 11
IsPerm3dObject, 54
IsPermPreXModWithObjects, 61
IsPreCat1Morphism, 24
IsPreCrossedSquare, 55
IsPreCrossedSquareMorphism, 56
IsPreXCat1Group, 16
IsPreXMod, 11
IsPreXModMorphism, 21
IsPreXModWithObjects, 61
IsSection, 38
IsSimplyConnected2DimensionalGroup, 31
IsSingleValued

for pre-xmod morphisms, 22
IsStemDomain

for crossed modules of groups, 35
for groups, 34

IsSurjective
for pre-xmod morphisms, 22

IsTotal
for pre-xmod morphisms, 22

IsTrivialAction2DimensionalGroup, 10
IsUp2DimensionalMapping, 37
IsXMod, 11
IsXModMorphism, 21
IsXModWithObjects, 61

Kernel
for 2d-mappings, 26

Kernel2DimensionalMapping, 26
KernelEmbedding, 14

Left2DimensionalGroup, 55
Left2DimensionalMorphism, 55
loop space, 63

LoopsXMod, 63
LowerCentralSeriesOfXMod, 31
LueXMod, 44

MappingToOne, 65
morphism, 21
morphism of 2d-group, 21
morphism of 3d-group, 55
MorphismOfInducedXMod, 48

Name, 53
for cat1-groups, 14
for crossed modules, 9

NaturalMorphismByNormalSubPreXMod, 27
NilpotencyClass2DimensionalGroup, 31
Normalizer, 30
NormalSubXMods, 11
NorrieXMod, 44

ObjectList, 61
operations on morphisms, 25
order of a 2d-automorphism, 22

Peiffer subgroup, 12
PeifferSubgroup, 12
PermAutomorphismAsXModMorphism, 43
pre-crossed module, 12
PreCat1Group, 13
PreCat1GroupByEndomorphisms, 13
PreCat1GroupByNormalSubgroup, 15
PreCat1GroupByTailHeadEmbedding, 13
PreCat1GroupOfPreXMod, 17
PreCat1Morphism, 24
PreCat1MorphismByHoms, 24
PreCat2Group, 58
PreCat2GroupByPreCat1Groups, 58
PreXModByBoundaryAndAction, 12
PreXModMorphism, 22
PreXModMorphismByHoms, 22
PreXModOfPreCat1Group, 17
PrincipalDerivation, 38
PrincipalDerivations, 42

Range, 55
for 2d-group mappings, 21
for cat1-groups, 14
for crossed modules, 9

RangeEmbedding, 14

RangeHom, 21
regular derivation, 37
RegularDerivations, 41
RegularSections, 42
restriction mapping, 65
ReverseCat1Group, 15
Right2DimensionalGroup, 55
Right2DimensionalMorphism, 56
Root2dGroup, 61

section, of cat1-group, 37
SectionByDerivation, 38
SectionByHomomorphism, 38
selection of a small cat1-group, 18
SinglePiecePreXModWithObjects, 60
Size

for cat1-groups, 14
for crossed modules, 9

SmallerDegreePerm2DimensionalDomain
for pre-cat1 morphisms, 24

Source, 55
for 2d-group mappings, 21
for cat1-groups, 14
for crossed modules, 9

SourceHom, 21
StabilizerSubgroupXMod, 30
StructureDescription

for 2d-groups, 20
SubPreXMod, 12
SubXMod, 11
SurjectiveInducedXMod, 48

TailMap, 14
Transpose3dGroup, 53
TrivialSubXMod, 11

up 2d-mapping of 2d-group, 37
Up2DimensionalGroup, 55
Up2DimensionalMorphism, 55
UpGeneratorImages, 37
UpImagePositions, 37

version 1 for GAP 3, 68
version 2.001 for GAP 4, 68

Whitehead group, 37
Whitehead monoid, 37
Whitehead multiplication, 37

WhiteheadGroupTable, 41
WhiteheadMonoidTable, 40
WhiteheadOrder, 39
WhiteheadPermGroup, 41
WhiteheadProduct, 39
WhiteheadTransformationMonoid, 40
WhiteheadXMod, 44

XMod, 8
XModAction

for crossed modules of groupoids, 61
for crossed modules of groups, 9

XModByAbelianModule, 8
XModByAutomorphismGroup, 8
XModByBoundaryAndAction, 8
XModByCentralExtension, 8
XModByGroupOfAutomorphisms, 8
XModByInnerAutomorphismGroup, 8
XModByNormalSubgroup, 8
XModByPeifferQuotient, 12
XModByTrivialAction, 8
XModCentre, 46
XModMorphism, 22
XModMorphismByHoms, 22
XModOfCat1Group, 17

